code:
@SOURCE_CHECK
#DIM 키스마크반감
#DIM 욕구불만도상승
#DIM 경험증가량
#DIM 체위보존
#DIM V섹스보존
#DIM A섹스보존
#DIM PLAYER체위보존
#DIM PLAYERV섹스보존
#DIM PLAYERA섹스보존
#DIMS ゴム
#DIM 중독보정
VARSET LOCAL
SELECTCOM = TFLAG:3
중독보정 = 0
ゴム =
;조교 SELECTCOM의 시간경과
CALL TRACHECK_TIME

;-------------------------------------------------
;몸 접촉의 계속, 해제
;-------------------------------------------------
CALL TOUCH_SUCCESSION(TARGET)

;-------------------------------------------------
;TRYCALLFORM TOUCH_SUCCESSION_{SELECTCOM}(TARGET)
;TOUCH_SUCCESSION로 가고 있는 처리를 각 커맨드 파일에 분산시키는 것으로, 커맨드 추가를 손쉽게 하고 싶다
;-------------------------------------------------

IF !카운터불발커맨드(SELECTCOM)
	;카운터
	FOR LOCAL, 1, GET_TARGETNUM() + 1
		SIF TARGET:LOCAL <= 0
			CONTINUE
		CALL EVENT_COUNTER(TARGET:LOCAL, LOCAL)
		CALL EVENT_COUNTER_SOURCE(TARGET:LOCAL, LOCAL)
	NEXT
	;CALL EVENT_COUNTER_COMBINATION
ENDIF
FOR LOCAL,0,CHARANUM
	SIF CFLAG:LOCAL:현재위치 != CFLAG:MASTER:현재위치
		CONTINUE
	;성교 플래그의 설정
	IF TEQUIP:LOCAL:V접촉부위 == 1
		TEQUIP:LOCAL:50 = MASTER
	ELSE
		TEQUIP:LOCAL:50 = -1
	ENDIF
	IF TEQUIP:LOCAL:A접촉부위 == 1
		TEQUIP:LOCAL:51 = MASTER
	ELSE
		TEQUIP:LOCAL:51 = -1
	ENDIF
	IF TEQUIP:LOCAL:P접촉부위 == 6
		TEQUIP:PLAYER:50 = LOCAL
	ELSE
		TEQUIP:PLAYER:50 = -1
	ENDIF
	IF TEQUIP:LOCAL:P접촉부위 == 7
		TEQUIP:PLAYER:51 = LOCAL
	ELSE
		TEQUIP:PLAYER:51 = -1
	ENDIF
NEXT

FOR LOCAL,0,CHARANUM
	SIF CFLAG:LOCAL:현재위치 != CFLAG:MASTER:현재위치
		CONTINUE
	;P가 들어 있지 않으면 체위 리셋
	IF TEQUIP:LOCAL:50 < 0 && TEQUIP:LOCAL:51 < 0
		TEQUIP:LOCAL:체위 = 0
	;V성교중 또한 전회 V성교라면 TEQUIP는 복귀
	ELSEIF TEQUIP:LOCAL:50 >= 0 && TEQUIP:LOCAL:116
		SIF TEQUIP:LOCAL:116 != 3	;指挿入でない場合のみ適用
			TEQUIP:LOCAL:106 = TEQUIP:LOCAL:116
	;A성교중 또한 전회 A성교라면 TEQUIP는 복귀
	ELSEIF TEQUIP:LOCAL:51 >= 0 && TEQUIP:LOCAL:117
		SIF TEQUIP:LOCAL:117 != 3	;指挿入でない場合のみ適用
			TEQUIP:LOCAL:107 = TEQUIP:LOCAL:117
	ENDIF
	;내친 김에 발기도의 보존
	CFLAG:LOCAL:발기도 = BASE:LOCAL:발기
NEXT
;-------------------------------------------------
;카운터 탈의의 열매 처리
;-------------------------------------------------
CALL DATUI_LATE

;-------------------------------------------------
;@COM의 호출
;-------------------------------------------------
;TARGET 마다 COMF 호출
CALL CALL_COM(TARGET)

체위보존 = TEQUIP:체위
V섹스보존 = TEQUIP:V섹스
A섹스보존 = TEQUIP:A섹스
PLAYER체위보존 = TEQUIP:MASTER:체위
PLAYERV섹스보존 = TEQUIP:PLAYER:V섹스
PLAYERA섹스보존 = TEQUIP:PLAYER:A섹스
TFLAG:전회커맨드결과 = RESULT
;RESULT가-1이라면 캔슬 판정
SIF TFLAG:전회커맨드결과 < 0
	RETURN 0
;RESULT가 0이라면 SOURCE 계산 SKIP
SIF !TFLAG:전회커맨드결과
	GOTO SOURCE_SKIP
;구상 내에서 소스 추가 처리
CALL KOJO_MESSAGE_SEND("EXTRASOURCE", SELECTCOM, TARGET)

IF OAZUKE(TARGET)
	PRINTFORML 미뤄짐을 당한 %타겟은% 불만스러운 듯 하다…
	SOURCE:반감 += 3000
	SOURCE:굴종 += 500
	SOURCE:공순 -= 1500
ENDIF

CALL PEARL_CHK_MASTER

FOR LOCAL,0,CHARANUM
	키스마크반감 = 0
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
	;로터삽입
	IF CFLAG:LOCAL:로터삽입
		SOURCE:LOCAL:쾌V += TIME_PROGRESS(TIME:1) * 10
		CFLAG:LOCAL:로터삽입 = MAX(CFLAG:LOCAL:로터삽입 - TIME_PROGRESS(TIME:1),0)
		IF CFLAG:LOCAL:로터삽입 == 0
			PRINTFORMW %CALLNAME:LOCAL%가 장착하고 있던 로터의 전지가 끊어졌다
			IF TALENT:LOCAL:처녀 < 1
				EXP:LOCAL:V경험 += 9
			ELSE
				TFLAG:2 = 1
			ENDIF
		ENDIF
	ENDIF
	IF CFLAG:LOCAL:로터A삽입
		SOURCE:LOCAL:쾌A += TIME_PROGRESS(TIME:1) * 10
		CFLAG:LOCAL:로터A삽입 = MAX(CFLAG:LOCAL:로터A삽입 - TIME_PROGRESS(TIME:1),0)
		IF CFLAG:LOCAL:로터A삽입 == 0
			PRINTFORMW %CALLNAME:LOCAL%가 장착하고 있던 로터의 전지가 끊어졌다
			EXP:LOCAL:A경험 += 9
		ENDIF
	ENDIF
	IF TFLAG:키스마크 && LOCAL > 0 && !TFLAG:붕대 && !FLAG:70 && !CFLAG:LOCAL:수면
		IF CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치 && TFLAG:키스마크 != LOCAL && !TFLAG:거품목욕
			PRINTFORMW %CALLNAME:LOCAL%는 %CALLNAME:MASTER%의 키스마크를 보고 비난했다…
			IF TALENT:LOCAL:연모
				키스마크반감 = 500
			ELSEIF TALENT:LOCAL:사모
				키스마크반감 = 300
			ENDIF
			키스마크반감 = 키스마크반감 * (TALENT:LOCAL:정조 + 2) / 2
			SOURCE:LOCAL:반감 += 키스마크반감
		ENDIF
	ENDIF
;입지 않으면 노출이 증가
	SIF !EQUIP:LOCAL:하반신속옷1 && !EQUIP:LOCAL:하반신속옷2 && !TALENT:LOCAL:음란
		SOURCE:LOCAL:노출 += TIME_PROGRESS(TIME:1)

	SIF IsInsenceValid(LOCAL) || IsGardenValid(LOCAL)
		SOURCE:LOCAL:욕정 += TIME_PROGRESS(TIME:1) * (RAND:((ABL:LOCAL:욕망) + 1) + 3)

	SIF TEQUIP:LOCAL:유두빨기
		CALL EQUIP_COM11(LOCAL)
	SIF TEQUIP:LOCAL:가슴주무르기
		CALL EQUIP_COM12(LOCAL)
	SIF TEQUIP:LOCAL:계속키스
		CALL EQUIP_COM19(LOCAL)
	;장착계는 다음에 생각한다
	SIF LOCAL > 0 && !CFLAG:LOCAL:동실
		CONTINUE
	;-------------------------------------------------
	;주인, 동성, 조교자의 능력의 처리
	;-------------------------------------------------
	CALL SOURCE_SEX_CHECK(LOCAL)
	CALL PLAYER_SKILL_CHECK(LOCAL)

	;-------------------------------------------------
	;기교, 호감도, 무드, 이성의 체크
	;-------------------------------------------------
	FOR LOCAL:1,0,100
		SOURCE:LOCAL:(LOCAL:1) = SOURCE:LOCAL:(LOCAL:1) *  MASTER_FAVOR_CHECK(LOCAL,LOCAL:1) * TECHNIQUE_CHECK(LOCAL,LOCAL:1) / 10000
		SOURCE:LOCAL:(LOCAL:1) = SOURCE:LOCAL:(LOCAL:1) *  MOOD_CHECK(LOCAL,LOCAL:1) * REASON_CHECK(LOCAL,LOCAL:1) / 10000
	NEXT

	;-------------------------------------------------
	;중독 충족의 체크
	;-------------------------------------------------
	;자위중독
	IF EXP_UP(22,LOCAL)
		FOR LOCAL:1,0,100
			;쾌감, 성행동, 욕정, 공순, 노출, 굴종
			SIF (LOCAL:1 >= 0 && LOCAL:1 <= 9) || LOCAL:1 == 11 || (LOCAL:1 >= 15 && LOCAL:1 <= 18)
				SOURCE:LOCAL:(LOCAL:1) = SOURCE:LOCAL:(LOCAL:1) * (100 + GET_REVISION(ABL:LOCAL:자위중독, 500, 15)) / 100
		NEXT
	ENDIF
	;-------------------------------------------------
	;바이브 등 붙이고 있을 뿐 아이템의 체크
	;-------------------------------------------------
	CALL EQUIP_CHECK(LOCAL)
	;-------------------------------------------------
	;경험 가산
	;-------------------------------------------------
	CALL SOURCE_EXP(LOCAL)
	;-------------------------------------------------
	;처녀, 동정상실의 처리
	;-------------------------------------------------
	CALL LOST_VIRGIN(LOCAL)
	CALL 동정키스미경험상실(LOCAL)
	;-------------------------------------------------
	;쾌C의 소스
	;-------------------------------------------------
	CALL SOURCE_쾌C(LOCAL)
	;-------------------------------------------------
	;쾌V의 소스
	;-------------------------------------------------
	CALL SOURCE_쾌V(LOCAL)
	;-------------------------------------------------
	;쾌A의 소스
	;-------------------------------------------------
	CALL SOURCE_쾌A(LOCAL)
	;-------------------------------------------------
	;쾌B의 소스
	;-------------------------------------------------
	CALL SOURCE_쾌B(LOCAL)
	;-------------------------------------------------
	;-------------------------------------------------
	;쾌M의 소스
	;-------------------------------------------------
	CALL SOURCE_쾌M(LOCAL)
	;-------------------------------------------------
	;여쾌C의 소스
	;-------------------------------------------------
	SIF !(TALENT:PLAYER:2 & 2)
	CALL SOURCE_여쾌C(LOCAL)
	;-------------------------------------------------
	;여쾌V의 소스
	;-------------------------------------------------
	CALL SOURCE_여쾌V(LOCAL)
	;-------------------------------------------------
	;여쾌A의 소스
	;-------------------------------------------------
	CALL SOURCE_여쾌A(LOCAL)
	;-------------------------------------------------
	;여쾌B의 소스
	;-------------------------------------------------
	CALL SOURCE_여쾌B(LOCAL)
	;-------------------------------------------------
	;여쾌M의 소스
	;-------------------------------------------------
	CALL SOURCE_여쾌M(LOCAL)
	;-------------------------------------------------
NEXT

;-------------------------------------------------
;같은 커맨드의 연속실행에 의한 상하의 처리(쾌락계)
;-------------------------------------------------
IF SELECTCOM == PREVCOM && TFLAG:50 == TFLAG:151 && (SELECTCOM < 350 || SELECTCOM >= 500) && 연속실행()
	;UP:0,1,2,3은 절정에 관련되므로 먼저 처리되어 있다
	CUP:TARGET:쾌C /= 2
	CUP:TARGET:쾌V /= 2
	CUP:TARGET:쾌A /= 2
	CUP:TARGET:쾌B /= 2
	CUP:TARGET:쾌M /= 2
ENDIF
IF TCVAR:TARGET:억지로
	TIMES CUP:TARGET:쾌C, 0.75
	TIMES CUP:TARGET:쾌V, 0.75
	TIMES CUP:TARGET:쾌A, 0.75
	TIMES CUP:TARGET:쾌B, 0.75
	TIMES CUP:TARGET:쾌M, 0.75
ENDIF
IF TEQUIP:TARGET:애태우기모드
	TIMES CUP:TARGET:쾌C, 0.90
	TIMES CUP:TARGET:쾌V, 0.90
	TIMES CUP:TARGET:쾌A, 0.90
	TIMES CUP:TARGET:쾌B, 0.90
	TIMES CUP:TARGET:쾌M, 0.90
ENDIF
IF TCVAR:TARGET:만취
	TIMES CUP:TARGET:쾌C, 0.70
	TIMES CUP:TARGET:쾌V, 0.70
	TIMES CUP:TARGET:쾌A, 0.70
	TIMES CUP:TARGET:쾌B, 0.70
	TIMES CUP:TARGET:쾌M, 0.70
ENDIF

FOR LOCAL,0,CHARANUM
	;장착계는 다음에 생각한다
	SIF LOCAL > 0 && !CFLAG:LOCAL:동실
		CONTINUE
	;그 외계 소스 계산 처리
	CALL SOURCE_CVAB_EXTRA(LOCAL)

	BASE:LOCAL:발기 = BASE:LOCAL:발기 + CUP:LOCAL:쾌C / 5 + CUP:LOCAL:쾌V / 10 + CUP:LOCAL:쾌A / 10 + CUP:LOCAL:쾌B / 10
	;-------------------------------------------------
	;절정 처리
	;-------------------------------------------------
	CALL ORGASM_ADD(LOCAL)
NEXT

;-------------------------------------------------
;사정 체크(양성·남자)
;-------------------------------------------------
;요점 검토
CALL EVENT_SHOOT

FOR LOCAL,0,CHARANUM
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
	;장착계는 다음에 생각한다
	IF LOCAL > 0 && !CFLAG:LOCAL:동실
		SIF IsGardenValid(LOCAL)
			CALL SOURCE_욕정(LOCAL)
		CONTINUE
	ENDIF
	;-------------------------------------------------
	;조교 대상의 분유체크
	;-------------------------------------------------
	CALL TARGET_MILK_CHECK(LOCAL)
	;-------------------------------------------------
	;중독 충족 체크 ;KR판 구잔재 또는 오리지날
	;-------------------------------------------------
	;정액중독, 질내사정중독, 항내사정중독
	IF FLAG:70 == 1 && CFLAG:TARGET:344 == 0
	ELSE
		IF EXP_UP(14, LOCAL)
			IF TALENT:MASTER:농후정액 == 1
				중독보정 = 100 + GET_REVISION(ABL:LOCAL:정액중독 * EXP_UP(14, LOCAL), 500, 25)
			ELSE
				중독보정 = 100 + GET_REVISION(ABL:LOCAL:정액중독 * EXP_UP(14, LOCAL), 500, 50)
			ENDIF
			SOURCE:LOCAL:사랑 = SOURCE:LOCAL:사랑 * 중독보정 / 100
			SOURCE:LOCAL:성행동 = SOURCE:LOCAL:성행동 * 중독보정 / 100
			SOURCE:LOCAL:달성 = SOURCE:LOCAL:달성 * 중독보정 / 100
			SOURCE:LOCAL:욕정 = SOURCE:LOCAL:욕정 * 중독보정 / 100
			SOURCE:LOCAL:공순 = SOURCE:LOCAL:공순 * 중독보정 / 100
			SOURCE:LOCAL:굴종 = SOURCE:LOCAL:굴종 * 중독보정 / 100
		ENDIF
		IF EXP_UP(17, LOCAL)
			IF TALENT:MASTER:농후정액 == 1
				중독보정 = 100 + GET_REVISION(ABL:LOCAL:항내사정중독 * EXP_UP(17, LOCAL), 500, 25)
			ELSE
				중독보정 = 100 + GET_REVISION(ABL:LOCAL:항내사정중독 * EXP_UP(17, LOCAL), 500, 50)
			ENDIF
			SIF TALENT:LOCAL:동물귀 && TEQUIP:LOCAL:체위 == 2
				중독보정 = 중독보정 * 11 / 10
			SOURCE:LOCAL:사랑 = SOURCE:LOCAL:사랑 * 중독보정 / 100
			SOURCE:LOCAL:성행동 = SOURCE:LOCAL:성행동 * 중독보정 / 100
			SOURCE:LOCAL:달성 = SOURCE:LOCAL:달성 * 중독보정 / 100
			SOURCE:LOCAL:욕정 = SOURCE:LOCAL:욕정 * 중독보정 / 100
			SOURCE:LOCAL:공순 = SOURCE:LOCAL:공순 * 중독보정 / 100
			SOURCE:LOCAL:굴종 = SOURCE:LOCAL:굴종 * 중독보정 / 100
		ENDIF
		IF EXP_UP(16, LOCAL)
			IF TALENT:MASTER:농후정액 == 1
				중독보정 = 100 + GET_REVISION(ABL:LOCAL:질내사정중독 * EXP_UP(16, LOCAL), 500, 25)
			ELSE
				중독보정 = 100 + GET_REVISION(ABL:LOCAL:질내사정중독 * EXP_UP(16, LOCAL), 500, 50)
			ENDIF
			SIF TALENT:LOCAL:동물귀 && TEQUIP:LOCAL:체위 == 2
				중독보정 = 중독보정 * 11 / 10
			SIF CFLAG:LOCAL:생리주기 == 7
				중독보정 = 중독보정 * 12 / 10
			SOURCE:LOCAL:사랑 = SOURCE:LOCAL:사랑 * 중독보정 / 100
			SOURCE:LOCAL:성행동 = SOURCE:LOCAL:성행동 * 중독보정 / 100
			SOURCE:LOCAL:달성 = SOURCE:LOCAL:달성 * 중독보정 / 100
			SOURCE:LOCAL:욕정 = SOURCE:LOCAL:욕정 * 중독보정 / 100
			SOURCE:LOCAL:공순 = SOURCE:LOCAL:공순 * 중독보정 / 100
			SOURCE:LOCAL:굴종 = SOURCE:LOCAL:굴종 * 중독보정 / 100
		ENDIF
	ENDIF
	;-------------------------------------------------
	;면간으로부터의 복귀 사전 처리
	;-------------------------------------------------
	CALL DRUNK_RECOVER_BEFORE(TARGET)
	;-------------------------------------------------
	;사랑의 소스
	;-------------------------------------------------
	CALL SOURCE_사랑(LOCAL)
	;-------------------------------------------------
	;성행동의 소스
	;-------------------------------------------------
	CALL SOURCE_성행동(LOCAL)
	;-------------------------------------------------
	;달성의 소스
	;-------------------------------------------------
	CALL SOURCE_달성(LOCAL)
	;-------------------------------------------------
	;아픔의 소스
	;-------------------------------------------------
	CALL SOURCE_고통(LOCAL)
	;-------------------------------------------------
	;우려의 소스
	;-------------------------------------------------
	CALL SOURCE_공포(LOCAL)
	;-------------------------------------------------
	;액체의 소스
	;-------------------------------------------------
	CALL SOURCE_액체(LOCAL)
	;-------------------------------------------------
	;욕정의 소스
	;-------------------------------------------------
	CALL SOURCE_욕정(LOCAL)
	;-------------------------------------------------
	;공순의 소스
	;-------------------------------------------------
	CALL SOURCE_공순(LOCAL)
	;-------------------------------------------------
	;노출의 소스
	;-------------------------------------------------
	CALL SOURCE_노출(LOCAL)
	;-------------------------------------------------
	;굴종의 소스
	;-------------------------------------------------
	CALL SOURCE_굴종(LOCAL)

	;-------------------------------------------------
	;환락의 소스
	;-------------------------------------------------
	CALL SOURCE_환락(LOCAL)
	;-------------------------------------------------
	;정복의 소스
	;-------------------------------------------------
	CALL SOURCE_정복(LOCAL)
	;-------------------------------------------------
	;수동의 소스
	;-------------------------------------------------
	CALL SOURCE_수동(LOCAL)
	;-------------------------------------------------
	;유혹의 소스
	;-------------------------------------------------
	CALL SOURCE_유혹(LOCAL)
	;-------------------------------------------------
	;송구스러운 소스
	;-------------------------------------------------
	CALL SOURCE_송구스러운(LOCAL)
	;-------------------------------------------------
	;도발의 소스
	;-------------------------------------------------
	CALL SOURCE_도발(LOCAL)
	;-------------------------------------------------
	;봉사의 소스
	;-------------------------------------------------
	CALL SOURCE_봉사(LOCAL)
	;-------------------------------------------------
	;강요의 소스
	;-------------------------------------------------
	CALL SOURCE_강요(LOCAL)
	;-------------------------------------------------
	;가학의 소스
	;-------------------------------------------------
	CALL SOURCE_가학(LOCAL)

	;-------------------------------------------------
	;불결의 소스
	;-------------------------------------------------
	CALL SOURCE_불결(LOCAL)
	;-------------------------------------------------
	;울굴의 소스
	;-------------------------------------------------
	CALL SOURCE_울굴(LOCAL)
	;-------------------------------------------------
	;일탈의 소스
	;-------------------------------------------------
	CALL SOURCE_일탈(LOCAL)
	;-------------------------------------------------
	;반감의 소스
	;-------------------------------------------------
	CALL SOURCE_반감(LOCAL)
	;-------------------------------------------------
	;소질등에 의한 상하의 처리
	;-------------------------------------------------
	CALL SOURCE_EXTRA(LOCAL)
	;-------------------------------------------------
	;수면심도의 감소 처리
	;-------------------------------------------------
	CALL 수면심도감소(LOCAL)
NEXT
;-------------------------------------------------
;같은 커맨드의 연속실행에 의한 상하의 처리
;-------------------------------------------------
IF SELECTCOM == PREVCOM && TFLAG:50 == TFLAG:151 && (SELECTCOM < 350 || SELECTCOM >= 500) && 연속실행()
	;UP:0,1,2는 절정에 관련되므로 먼저 처리되어 있다
	UP:욕정 /= 2
	UP:공순 /= 2
	UP:굴복 /= 2
	UP:습득 /= 2
	UP:치정 /= 2
	UP:공포 /= 2
	UP:고통 /= 2
	UP:호의 /= 2
	UP:우월 /= 2
ENDIF

LOCAL:1 = 0
FOR LOCAL,0,CHARANUM
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
	LOCAL:1 += CUP:LOCAL:욕정 / 5
	LOCAL:1 += MAX(5,(ABL:LOCAL:마조끼 - 3)) * (CUP:LOCAL:고통 + CUP:LOCAL:고통) / 2
	LOCAL:1 += MAX(5,(ABL:LOCAL:마조끼 - 1)) * CUP:LOCAL:치정 / 2
	LOCAL:1 = MIN(500,LOCAL:1)
	BASE:LOCAL:발기 = BASE:LOCAL:발기 + LOCAL:1
	;장착계는 다음에 생각한다
	SIF LOCAL > 0 && !CFLAG:LOCAL:동실
		CONTINUE
	;-------------------------------------------------
	;체력·기력의 감소
	;-------------------------------------------------
	CALL SOURCE_DOWNBASE(LOCAL)
	;-------------------------------------------------
	;고통쾌락경험, 봉사쾌락경험의 체크
	;-------------------------------------------------
	CALL EXP_GOT_CHECK(LOCAL)
NEXT

$SOURCE_SKIP
;-------------------------------------------------
;조교 말
;-------------------------------------------------
;CALL KOJO_MESSAGE_COM
CALL KOJO_MESSAGE_SEND("COMMAND",-1,TARGET,-1,-1)
;삽화 정면계 고무 있는 사정시, 체위도 없어지는 것은 어째서다
IF V섹스보존 != -1 && GROUPMATCH(체위보존,1,2,3,4,5) && !FLAG:삽입애니메이션필터 & 아니메필터_삽입
	SIF FLAG:화상표시 && FLAG:삽화표시
		PRINTL
	SELECTCASE 체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_정상위("正常位_挿入")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_후배위("後背位_挿入")
	ENDSELECT
ENDIF

IF PLAYERV섹스보존 != -1 && GROUPMATCH(PLAYER체위보존, 1,2,3,4,5) && !FLAG:삽입애니메이션필터 & 아니메필터_삽입
	SIF FLAG:화상표시 && FLAG:삽화표시
		PRINTL
	SELECTCASE PLAYER체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_정상위("正常位_挿入")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_후배위("後背位_挿入")
	ENDSELECT
ENDIF

IF A섹스보존 != -1 && GROUPMATCH(체위보존,1,2,3,4,5) && !FLAG:삽입애니메이션필터 & 아니메필터_삽입
	SIF FLAG:화상표시 && FLAG:삽화표시
		PRINTL
	SELECTCASE 체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_정상위("Ａ正常位_挿入")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_후배위("Ａ後背位_挿入")
	ENDSELECT
ENDIF
IF PLAYERA섹스보존 != -1 && GROUPMATCH(PLAYER체위보존,1,2,3,4,5) && !FLAG:삽입애니메이션필터 & 아니메필터_삽입
	SIF FLAG:화상표시 && FLAG:삽화표시
		PRINTL
	SELECTCASE PLAYER체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_정상위("Ａ正常位_挿入" )
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_후배위("Ａ後背位_挿入" )
	ENDSELECT
ENDIF

;카운터
IF TARGET
	LOCAL:4 = TARGET
	DRAWLINE
	FOR LOCAL, 1, GET_TARGETNUM() + 1
		SIF TCVAR:(TARGET:LOCAL):30 > 1
			CONTINUE
		SIF TCVAR:(TARGET:LOCAL):20 && TFLAG:193 != 99 && !CFLAG:TARGET:동침중
			CALL KOJO_MESSAGE_SEND("COUNTER",2,TARGET:LOCAL,0,0)
	NEXT
	TARGET = LOCAL:4
ENDIF

;-------------------------------------------------
;수간으로부터의 복귀 처리
;-------------------------------------------------
CALL DRUNK_RECOVER(TARGET)

;-------------------------------------------------
;EXP 취득 표시
;-------------------------------------------------
LOCAL:2 = 0
LOCAL:3 = 0
FOR LOCAL,0,CHARANUM
	SIF CFLAG:LOCAL:현재위치 != CFLAG:MASTER:현재위치
		CONTINUE
	SIF TCVAR:LOCAL:표시컷
		CONTINUE
	FOR LOCAL:1,0,100
		IF EXP_UP(LOCAL:1,LOCAL)
			경험증가량 = EXP_UP(LOCAL:1,LOCAL)
			LOCAL:2 ++
			LOCAL:3 ++
			SIF LOCAL:2 == 1
				PRINTL
			IF LOCAL:1 == FLAG:가호 && FLAG:가호 && LOCAL == MASTER
				EXP:MASTER:(LOCAL:1) ++
				경험증가량 ++
			ENDIF
			PRINTFORML %EXPNAME:(LOCAL:1)% +{경험증가량}(%CALLNAME:LOCAL%)
			;당신은 무자각경험을 얻지 않게
			IF LOCAL:1 >= 0 && LOCAL:1 <= 3 && LOCAL && (FLAG:70 || CFLAG:장난)
				EXP:LOCAL:((LOCAL:1) + 60) += 1
			ENDIF
		ENDIF
	NEXT
	LOCAL:2 = 0
NEXT
SIF LOCAL:3
	WAIT
;-------------------------------------------------
;ABLUP
;-------------------------------------------------
IF !FLAG:ABLUP생략
	FOR LOCAL,0,CHARANUM
		SIF CFLAG:LOCAL:현재위치 != CFLAG:MASTER:현재위치
			CONTINUE
		SIF TCVAR:LOCAL:표시컷
			CONTINUE
		CALL SOURCE_ABLUP(LOCAL)
	NEXT
ENDIF
;-------------------------------------------------
;촉수의 사정시의 말
;-------------------------------------------------
;DEBUG	CALL KOJO_MESSAGE_PALAMCNG_E
CALL KOJO_MESSAGE_SEND("PALAM",5,TARGET,-1,-1)
;-------------------------------------------------
;노예의 절정시의 말
;-------------------------------------------------
; CALL MESSAGE_PALAMCNG_B2

CALL KOJO_MESSAGE_SEND("PALAM",2,TARGET,-1,-1)
;삽화 삽입∨절정
IF NOWEX:V절정 && V섹스보존 != -1 && GROUPMATCH(체위보존, 1,2,3,4,5)
	IF !GETBIT(TCVAR:50,1)
		SELECTCASE 체위보존
			CASE 1,3,4
				CALL PRINT_IMAGE_ANIMATION_정상위("正常位_絶頂")
			CASE 2,5
				CALL PRINT_IMAGE_ANIMATION_후배위("後背位_絶頂")
		ENDSELECT
	ENDIF
ENDIF
IF NOWEX:A절정 && A섹스보존 != -1 && GROUPMATCH(체위보존,1,2,3,4,5)
	IF !GETBIT(TCVAR:50,2)
		SELECTCASE 체위보존
			CASE 1,3,4
				CALL PRINT_IMAGE_ANIMATION_정상위("Ａ正常位_絶頂")
			CASE 2,5
				CALL PRINT_IMAGE_ANIMATION_후배위("Ａ後背位_絶頂")
		ENDSELECT
	ENDIF
ENDIF

;挿絵　PALYER삽입∨절정
IF NOWEX:MASTER:V절정 && PLAYERV섹스보존 != -1 && GROUPMATCH(PLAYER체위보존,1,2,3,4,5)
	IF NOWEX:TARGET:사정 == 0
		SELECTCASE PLAYER체위보존
			CASE 1,3,4
				CALL PRINT_IMAGE_ANIMATION_정상위("正常位_絶頂")
			CASE 2,5
				CALL PRINT_IMAGE_ANIMATION_후배위("後背位_絶頂")
		ENDSELECT
	ENDIF
ENDIF
IF NOWEX:MASTER:A절정 && PLAYERA섹스보존 != -1 && GROUPMATCH(PLAYER체위보존,1,2,3,4,5)
	IF NOWEX:TARGET:사정 == 0
		SELECTCASE PLAYER체위보존
			CASE 1,3,4
				CALL PRINT_IMAGE_ANIMATION_정상위("Ａ正常位_絶頂")
			CASE 2,5
				CALL PRINT_IMAGE_ANIMATION_후배위("Ａ後背位_絶頂")
		ENDSELECT
	ENDIF
ENDIF

;DEBUG	CALL KOJO_MESSAGE_PALAMCNG_B

;-------------------------------------------------
;노예의 초조하게 해 때의 말
;-------------------------------------------------
IF (EX:TARGET:C절정직전 + EX:TARGET:V절정직전 + EX:TARGET:A절정직전 + EX:TARGET:M절정직전 + EX:TARGET:B절정직전) > 0 && TEQUIP:TARGET:애태우기모드
	CALL KOJO_MESSAGE_SEND("PALAM",6,TARGET,-1,-1)
	;우후후시 계 장비 리셋
	FOR LOCAL,10, 80
		SELECTCASE LOCAL
			CASE 18 TO 20, 25 TO 40, 49 TO 51
				CONTINUE
		ENDSELECT
		TEQUIP:TARGET:LOCAL = 0
	NEXT
ENDIF
;-------------------------------------------------
;조교자의 사정시의 말
;-------------------------------------------------
CALL 사정절정표기(MASTER)
CALL 사정절정표기(TARGET)
CALL MESSAGE_PALAMCNG_B2

;DEBUG	CALL KOJO_MESSAGE_PALAMCNG_A
CALL KOJO_MESSAGE_SEND("PALAM",1,TARGET,-1,-1)

;삽화 삽입 사정 정상위 계열
IF GETBIT(TCVAR:50,1) && TEQUIP:V섹스 != -1 && !TCVAR:PLAYER:콘돔 && GROUPMATCH(체위보존,1,2,3,4,5)
	SIF FLAG:화상표시 && FLAG:삽화표시
		PRINTL
	SELECTCASE 체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_정상위("正常位_射精")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_후배위("後背位_射精")
	ENDSELECT
ENDIF
IF GETBIT(TCVAR:50,2) && TEQUIP:A섹스 != -1 && !TCVAR:PLAYER:콘돔 && GROUPMATCH(체위보존,1,2,3,4,5)
	SIF FLAG:화상표시 && FLAG:삽화표시
		PRINTL
	SELECTCASE 체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_정상위("Ａ正常位_射精")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_후배위("Ａ後背位_射精")
	ENDSELECT
ENDIF

;삽화 삽입 사정
;相手がコンドーム付けることあるのかな？
IF NOWEX:TARGET:사정 > 0 && PLAYERV섹스보존 != -1 && !TCVAR:TARGET:콘돔 && GROUPMATCH(PLAYER체위보존,1,2,3,4,5)
	SIF FLAG:화상표시 && FLAG:삽화표시
		PRINTL
	SELECTCASE PLAYER체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_정상위("正常位_射精")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_후배위("後背位_射精", 1)
	ENDSELECT
ENDIF
IF NOWEX:TARGET:사정 > 0 && PLAYERA섹스보존 != -1 && !TCVAR:TARGET:콘돔 && GROUPMATCH(PLAYER체위보존,1,2,3,4,5)
	SIF FLAG:화상표시 && FLAG:삽화표시
		PRINTL
	SELECTCASE PLAYER체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_정상위("Ａ正常位_射精")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_후배위("Ａ後背位_射精", 1)
	ENDSELECT
ENDIF

;-------------------------------------------------
;조교 후의 말
;-------------------------------------------------
;DEBUG	CALL KOJO_MESSAGE_PALAMCNG_C
CALL KOJO_MESSAGE_SEND("PALAM",3,TARGET,-1,-1)
;-------------------------------------------------
;파라미터 변동에 의한 말
;-------------------------------------------------
;DEBUG	CALL KOJO_MESSAGE_PALAMCNG_D
CALL KOJO_MESSAGE_SEND("PALAM",4,TARGET,-1,-1)
;-------------------------------------------------
;각인 취득의 체크(장난중은 금지)
;-------------------------------------------------
IF !CFLAG:TARGET:장난
	CALL MARK_GOT_CHECK(TARGET)
;-------------------------------------------------
;각인 취득에 의한 말(장난중은 금지)
;-------------------------------------------------
;DEBUG	CALL KOJO_MESSAGE_MARKCNG
	CALL KOJO_MESSAGE_SEND("MARK",-1,TARGET,-1,-1)
;-------------------------------------------------
;분유메시지
;-------------------------------------------------
SIF NOWEX:분유
	CALL TARGET_MILK_MESSAGE(TARGET)
;-------------------------------------------------
;순종 상승에 의한[저항]의 소멸 체크(장난중은 금지)
;-------------------------------------------------
	CALL JUJUN_UP_CHECK
ENDIF
RESETCOLOR
;-------------------------------------------------
;조교 소스의 표시
;-------------------------------------------------
CALL SHOW_SOURCE
;-------------------------------------------------
;수치 변동의 표시
;-------------------------------------------------
SETCOLOR C_YELLOW
;궁합
R = NO:PLAYER
SIF RELATION:R != 0
	PRINTFORML ＜궁합{RELATION:R/100}.{RELATION:R%100}배＞

SIF TCVAR:TARGET:마음대로해
	PRINTFORML ＜농락중＞

SIF SELECTCOM == 21
	PRINTFORML ＜【아무것도 하지 않는】의 효과로 %CALLNAME:MASTER%의 쾌입수량 증가＞
SIF SELECTCOM == 22
	PRINTFORML ＜【권해 받는】의 효과로 우월 증가＞

SIF TCVAR:커맨드C보정 && SOURCE:쾌C
	PRINTFORML ＜커맨드 보정에 의해 쾌C{ABS(TCVAR:커맨드C보정)}\%\@ TCVAR:커맨드C보정 > 0 ?  업# 다운\@＞
SIF TCVAR:커맨드B보정 && SOURCE:쾌B
	PRINTFORML ＜커맨드 보정에 의해 쾌B{ABS(TCVAR:커맨드B보정)}\%\@ TCVAR:커맨드B보정 > 0 ?  업# 다운\@＞
SIF TCVAR:커맨드V보정 && SOURCE:쾌V
	PRINTFORML ＜커맨드 보정에 의해 쾌V{ABS(TCVAR:커맨드V보정)}\%\@ TCVAR:커맨드V보정 > 0 ?  업# 다운\@＞
SIF TCVAR:커맨드A보정 && SOURCE:쾌A
	PRINTFORML ＜커맨드 보정에 의해 쾌A{ABS(TCVAR:커맨드A보정)}\%\@ TCVAR:커맨드A보정 > 0 ?  업# 다운\@＞
SIF TCVAR:커맨드M보정 && SOURCE:쾌M
	PRINTFORML ＜커맨드 보정에 의해 쾌M{ABS(TCVAR:커맨드M보정)}\%\@ TCVAR:커맨드M보정 > 0 ?  업# 다운\@＞
SIF SELECTCOM == PREVCOM && TFLAG:50 == TFLAG:151 && (SELECTCOM < 350 || SELECTCOM >= 500) && 연속실행()
	PRINTL ＜동일 커맨드 연속실행＞
;기력 0
SIF BASE:1 <= 0
	PRINTL ★기력 0★
RESETCOLOR
LOCAL:1 = 0
FOR LOCAL,0,CHARANUM
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
	IF DOWNBASE:LOCAL:체력
		IF DOWNBASE:LOCAL:체력 > 0
			SETCOLOR C_YELLOW
			SIF DOWNBASE:LOCAL:체력 > 150
				SETCOLOR C_RED
			PRINTFORM 체력-{DOWNBASE:LOCAL:체력}
		ELSE
			SETCOLOR C_GREEN
			PRINTFORM 체력+{-DOWNBASE:LOCAL:체력}
		ENDIF
		LOCAL:1 ++
	ENDIF
	IF DOWNBASE:LOCAL:기력
		IF DOWNBASE:LOCAL:기력 > 0
			SETCOLOR C_YELLOW
			SIF DOWNBASE:LOCAL:기력 > 150
				SETCOLOR C_RED
			PRINTFORM 기력-{DOWNBASE:LOCAL:기력}
		ELSE
			SETCOLOR C_GREEN
			PRINTFORM 기력+{-DOWNBASE:LOCAL:기력}
		ENDIF
		LOCAL:1 ++
	ENDIF
	RESETCOLOR
	SIF LOCAL:1
		PRINTFORML (%CALLNAME:LOCAL%)
	LOCAL:1 = 0
NEXT
SIF TIME_PROGRESS(TIME:1) > 0 && SELECTCOM != 400 && SELECTCOM != 402 && SELECTCOM != 604 && !CFLAG:MASTER:우후후
	PRINTFORML +{TIME_PROGRESS(TIME:1)}분
SIF TIME_PROGRESS(TIME:1) >= 40
	TCVAR:TARGET:가능한한움직이지않는다 = 1
PRINTL

;-------------------------------------------------
;파라미터의 상승＆표시(DOWN도 여기서)
;-------------------------------------------------
VARSET LOCAL
FOR LOCAL:1,0,CHARANUM
	SIF TCVAR:표시컷
		CONTINUE
	LOCAL:3 = 0
	FOR LOCAL,0,100
		SIF PALAMNAME:LOCAL == ""
			CONTINUE
		SIF CUP:(LOCAL:1):LOCAL - CDOWN:(LOCAL:1):LOCAL == 0
			CONTINUE
		IF !LOCAL:3
			PRINTL
			PRINTFORML %CALLNAME:(LOCAL:1)%
		ENDIF
		LOCAL:3 ++
		LOCAL:2 = CUP:(LOCAL:1):LOCAL - CDOWN:(LOCAL:1):LOCAL
		SIF CFLAG:(LOCAL:1):현재위치 == CFLAG:MASTER:현재위치
			PRINTFORML %PALAMNAME:LOCAL,8,LEFT% {PALAM:(LOCAL:1):LOCAL,10,LEFT} + {CUP:(LOCAL:1):LOCAL,10,LEFT} =   {PALAM:(LOCAL:1):LOCAL + CUP:(LOCAL:1):LOCAL,10,LEFT}
		PALAM:(LOCAL:1):LOCAL += CUP:(LOCAL:1):LOCAL - CDOWN:(LOCAL:1):LOCAL
		CUP:(LOCAL:1):LOCAL = 0
		CDOWN:(LOCAL:1):LOCAL = 0
	NEXT
NEXT
PRINTL

IF GETBIT(TCVAR:50,1) && TEQUIP:V섹스 != -1 && !TCVAR:PLAYER:콘돔 && GROUPMATCH(체위보존,1,2,3,4,5)
	SELECTCASE 체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_사정후("正常位_中出し")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_사정후("後背位_中出し")
	ENDSELECT
	IF FLAG:화상표시 && FLAG:삽화표시
		WAIT
		PRINTL
	ENDIF
ENDIF

IF NOWEX:TARGET:사정 > 0 && PLAYERV섹스보존 != -1 && !TCVAR:TARGET:콘돔 && GROUPMATCH(PLAYER체위보존,1,2,3,4,5)
	SELECTCASE PLAYER체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_사정후("正常位_中出し")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_사정후("後背位_中出し", 1)
	ENDSELECT
	IF FLAG:화상표시 && FLAG:삽화표시
		WAIT
		PRINTL
	ENDIF
ENDIF

IF GETBIT(TCVAR:50,2) && TEQUIP:A섹스 != -1 && !TCVAR:PLAYER:콘돔 && GROUPMATCH(체위보존,1,2,3,4,5)
	SELECTCASE 체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_사정후("Ａ正常位_中出し")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_사정후("Ａ後背位_中出し")
	ENDSELECT
	IF FLAG:화상표시 && FLAG:삽화표시
		WAIT
		PRINTL
	ENDIF
ENDIF

IF NOWEX:TARGET:사정 > 0 && PLAYERA섹스보존 != -1 && !TCVAR:TARGET:콘돔 && GROUPMATCH(PLAYER체위보존,1,2,3,4,5)
	SELECTCASE PLAYER체위보존
		CASE 1,3,4
			CALL PRINT_IMAGE_ANIMATION_사정후("Ａ正常位_中出し")
		CASE 2,5
			CALL PRINT_IMAGE_ANIMATION_사정후("Ａ後背位_中出し", 1)
	ENDSELECT
	IF FLAG:화상표시 && FLAG:삽화표시
		WAIT
		PRINTL
	ENDIF
ENDIF
DRAWLINE
;-------------------------------------------------
;욕구불만도상승
;-------------------------------------------------
IF TCVAR:TARGET:욕구불만도관리 < GETPALAMLV(PALAM:욕정,15) && TARGET

	IF TCVAR:TARGET:미약
		욕구불만도상승 = (GETPALAMLV(PALAM:욕정,15) - TCVAR:TARGET:욕구불만도관리) * (30 + TCVAR:TARGET:발정 * 20 + TCVAR:TARGET:미약 * 20)
	ELSE
		욕구불만도상승 = (GETPALAMLV(PALAM:욕정,15) - TCVAR:TARGET:욕구불만도관리) * (30 + TCVAR:TARGET:발정 * 20 + TFLAG:선향옥토 * 10)
	ENDIF
	욕구불만도상승 = 욕구불만소질보정(TARGET, 욕구불만도상승)
	SIF TALENT:TARGET:스침 == 1
		욕구불만도상승 *= 3
	CFLAG:TARGET:욕구불만도 = LIMIT(욕구불만도상승 + CFLAG:TARGET:욕구불만도, 0, 1000)
	TCVAR:TARGET:욕구불만도관리 = GETPALAMLV(PALAM:욕정,15)
ENDIF
;-------------------------------------------------
;호감도 상승
;-------------------------------------------------
CALL FAVOR_CALC(TARGET)

;-------------------------------------------------
;신뢰도 상승(우후후 이외＆상승 대상 커맨드 실행시)
;-------------------------------------------------
SIF (TFLAG:98 && !CFLAG:TARGET:우후후) && !CFLAG:TARGET:수면
	CALL TRUST_CALC(TARGET)

;-------------------------------------------------
;泡風呂タイムアップ処理
;-------------------------------------------------
SIF TFLAG:거품목욕 && (DAY * 1440 + TIME >= TFLAG:거품목욕)
	CALL 泡風呂終了判定("タイムアップ")

;-------------------------------------------------
;잠복율の증감처리
;-------------------------------------------------
SIF GETBIT(FLAG:잠복,0)
	CALL 잠복율증감처리

;-------------------------------------------------
;수면심도 0에 의한 각성 처리
;-------------------------------------------------
SIF CFLAG:MASTER:장난 == 2 && (TCVAR:TARGET:수면심도 <= 0 || !수면시간(TARGET))
	CALL 수면간발각(TARGET)

;-------------------------------------------------
;껴안기모드의 해제
;-------------------------------------------------
IF TFLAG:102 == 3 && TFLAG:껴안기모드
	TFLAG:껴안기해방포인트 --
	IF TFLAG:껴안기해방포인트 <= 0
		PRINTFORML %조사처리(CALLNAME,"는")% 만족해 %CALLNAME:MASTER%의 몸으로부터 멀어졌다…
		CALL 덮쳐지기종료
		TFLAG:껴안기모드 = 0
		TCVAR:껴안기금지시간 = 90 + MAX(10 - ABL:욕망, 0) * 3 - MAX((CFLAG:적극성 + CFLAG:수줍음) * 5, 30)
	ENDIF
ENDIF

;-------------------------------------------------
;변수의 리셋
;-------------------------------------------------
CVARSET TCVAR, GETNUM(TCVAR, "봉사플래그"), 0
CVARSET TCVAR, GETNUM(TCVAR, "사정게이지증가량"), 0
CVARSET TCVAR, GETNUM(TCVAR, "만취간들킴"), 0
CVARSET TCVAR, GETNUM(TCVAR, "요리평가치"), 0
;-------------------------------------------------
;조교 소스의 표시(DOWN도 여기서)
;-------------------------------------------------
@SHOW_SOURCE
SIF TCVAR:표시컷
	RETURN 0
;PRINTL
SIF SOURCE:쾌C > 0
	PRINTFORM 쾌C({SOURCE:쾌C})
SIF SOURCE:쾌V > 0
	PRINTFORM 쾌V({SOURCE:쾌V})
SIF SOURCE:쾌A > 0
	PRINTFORM 쾌A({SOURCE:쾌A})
SIF SOURCE:쾌B > 0
	PRINTFORM 쾌B({SOURCE:쾌B})
SIF SOURCE:쾌M > 0
	PRINTFORM 쾌M({SOURCE:쾌M})
SIF SOURCE:사랑 > 0
	PRINTFORM 사랑({SOURCE:사랑})
SIF SOURCE:성행동 > 0
	PRINTFORM 성행동({SOURCE:성행동})
SIF SOURCE:달성 > 0
	PRINTFORM 달성감({SOURCE:달성})
SIF SOURCE:고통 > 0
	PRINTFORM 아픔({SOURCE:고통})
SIF SOURCE:공포 > 0
	PRINTFORM 공포({SOURCE:공포})
SIF SOURCE:액체 > 0
	PRINTFORM 액체({SOURCE:액체})
SIF SOURCE:욕정 > 0
	PRINTFORM 욕정({SOURCE:욕정})
SIF SOURCE:공순 > 0
	PRINTFORM 공순({SOURCE:공순})
SIF SOURCE:노출 > 0
	PRINTFORM 노출({SOURCE:노출})
SIF SOURCE:굴종 > 0
	PRINTFORM 굴복({SOURCE:굴종})
SIF SOURCE:환락 > 0
	PRINTFORM 환락({SOURCE:환락})
SIF SOURCE:정복 > 0
	PRINTFORM 정복({SOURCE:정복})
SIF SOURCE:강요 > 0
	PRINTFORM 정복({SOURCE:강요})
SIF SOURCE:수동 > 0
	PRINTFORM 수동({SOURCE:수동})
SIF SOURCE:불결 > 0
	PRINTFORM 불결({SOURCE:불결})
SIF SOURCE:울굴 > 0
	PRINTFORM 울굴({SOURCE:울굴})
SIF SOURCE:일탈 > 0
	PRINTFORM 일탈({SOURCE:일탈})
SIF SOURCE:반감 > 0
	PRINTFORM 반감({SOURCE:반감})
PRINTL

@OAZUKE(ARG)
#FUNCTION
SIF CFLAG:ARG:우후후 != 2
	RETURNF 0
SELECTCASE TCVAR:ARG:20
	CASE 41
	;정상위 V
		SIF SELECTCOM != 320
			RETURNF 1
	CASE 43
	;백 V
		SIF SELECTCOM != 322
			RETURNF 1
	CASE 44
	;백 A
		SIF SELECTCOM != 323
			RETURNF 1
ENDSELECT
RETURNF 0


code:
;---------------------------------------------------
;마이나스 SOURCE
;---------------------------------------------------
;FileName_SOURCE1.ERB ------------------------------ Rev1. 00
;불결의 소스 처리
;CALL		USER
;ARG		캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@SOURCE_불결(ARG)
VARSET LOCAL
LOCAL = SOURCE:ARG:불결

;ABL:순종을 본다
SELECTCASE ABL:ARG:순종
	CASE 0
		TIMES LOCAL , 1.50
	CASE 1 TO 3
		TIMES LOCAL , 1.30
	CASE 4 TO 6
		TIMES LOCAL , 1.00
	CASE 7 TO 9
		TIMES LOCAL , 0.75
	CASE 10 TO 12
		TIMES LOCAL , 0.50
	CASE 13 TO 15
		TIMES LOCAL , 0.25
	CASEELSE
		TIMES LOCAL , 0.10
ENDSELECT
CUP:ARG:불쾌 += LOCAL

;FileName_SOURCE1.ERB ------------------------------ Rev1. 00
;울굴의 소스 처리
;CALL		USER
;ARG		캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@SOURCE_울굴(ARG)
VARSET LOCAL
LOCAL = SOURCE:ARG:울굴
;기분
LOCAL = LOCAL * (10 - 2 * TALENT:ARG:15) / 10
;함락 소질
IF TALENT:ARG:연모
	TIMES LOCAL , 0.25
ELSEIF TALENT:ARG:사모
	TIMES LOCAL , 0.75
ENDIF

;ABL:순종을 본다
SELECTCASE ABL:ARG:순종
	CASE 0
		TIMES LOCAL , 1.50
	CASE 1 TO 3
		TIMES LOCAL , 1.30
	CASE 4 TO 6
		TIMES LOCAL , 1.00
	CASE 7 TO 9
		TIMES LOCAL , 0.75
	CASE 10 TO 12
		TIMES LOCAL , 0.50
	CASE 13 TO 15
		TIMES LOCAL , 0.25
	CASEELSE
		TIMES LOCAL , 0.10
ENDSELECT
;ABL:욕망을 본다
SELECTCASE ABL:ARG:욕망
	CASE 0
		TIMES LOCAL , 2.00
	CASE 1 TO 3
		TIMES LOCAL , 1.50
	CASE 4 TO 6
		TIMES LOCAL , 1.00
	CASE 7 TO 9
		TIMES LOCAL , 0.75
	CASE 10 TO 12
		TIMES LOCAL , 0.50
	CASE 13 TO 15
		TIMES LOCAL , 0.25
	CASEELSE
		TIMES LOCAL , 0.10
ENDSELECT
CUP:ARG:억울 += LOCAL

;FileName_SOURCE1.ERB ------------------------------ Rev1. 00
;일탈의 소스 처리
;CALL		USER
;ARG		캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@SOURCE_일탈(ARG)
VARSET LOCAL
;호기심과 보수적은 여기서 처리
SIF TALENT:ARG:성적흥미 > 0
	TIMES SOURCE:ARG:일탈 , 0.80
SIF TALENT:ARG:성적흥미 < 0
	TIMES SOURCE:ARG:일탈 , 2.00

LOCAL = SOURCE:ARG:일탈

;ABL:순종을 본다
SELECTCASE ABL:ARG:순종
	CASE 0
		TIMES LOCAL , 1.50
	CASE 1 TO 3
		TIMES LOCAL , 1.30
	CASE 4 TO 6
		TIMES LOCAL , 1.00
	CASE 7 TO 9
		TIMES LOCAL , 0.75
	CASE 10 TO 12
		TIMES LOCAL , 0.50
	CASE 13 TO 15
		TIMES LOCAL , 0.25
	CASEELSE
		TIMES LOCAL , 0.10
ENDSELECT

;ABL:마조끼를 본다
SELECTCASE ABL:ARG:마조끼
	CASE 0
		TIMES LOCAL , 1.00
	CASE 1 TO 3
		TIMES LOCAL , 0.90
	CASE 3 TO 5
		TIMES LOCAL , 0.75
	CASE 6,7
		TIMES LOCAL , 0.60
	CASE 8,9
		TIMES LOCAL , 0.50
	CASE 10 TO 12
		TIMES LOCAL , 0.30
	CASE 13 TO 15
		TIMES LOCAL , 0.20
	CASEELSE
		TIMES LOCAL , 0.10
ENDSELECT
CUP:ARG:반감 += LOCAL

;FileName_SOURCE1.ERB ------------------------------ Rev1. 00
;반감의 소스 처리
;CALL		USER
;ARG		캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@SOURCE_반감(ARG)
VARSET LOCAL
LOCAL = SOURCE:ARG:반감
;기분
LOCAL = LOCAL * (10 - 3 * TALENT:ARG:15) / 10
;상대 주도
SIF ARG != MASTER && CFLAG:ARG:우후후 == 2
	TIMES LOCAL, 0.3
SIF TALENT:ARG:스침 == 1
	LOCAL /= 2
SIF CFLAG:ARG:몸목적 && TFLAG:이성깎기
	LOCAL /= 2
;ABL:순종을 본다
LOCAL:1 = ABL:ARG:순종 + ABL:ARG:친밀
SELECTCASE LOCAL:1
	CASE 0
		TIMES LOCAL , 2.00
	CASE 1 TO 3
		TIMES LOCAL , 1.25
	CASE 4 TO 7
		TIMES LOCAL , 1.00
	CASE 8 TO 10
		TIMES LOCAL , 0.75
	CASE 11 TO 14
		TIMES LOCAL , 0.50
	CASE 15 TO 19
		TIMES LOCAL , 0.25
	CASE 20 TO 24
		TIMES LOCAL , 0.10
	CASEELSE
		TIMES LOCAL , 0.05
ENDSELECT

;ABL:봉사정신을 본다
SELECTCASE ABL:ARG:봉사정신
	CASE 0
		TIMES LOCAL , 1.00
	CASE 1 TO 3
		TIMES LOCAL , 0.90
	CASE 3 TO 5
		TIMES LOCAL , 0.75
	CASE 6,7
		TIMES LOCAL , 0.60
	CASE 8,9
		TIMES LOCAL , 0.50
	CASE 10 TO 12
		TIMES LOCAL , 0.30
	CASE 13 TO 15
		TIMES LOCAL , 0.20
	CASEELSE
		TIMES LOCAL , 0.10
ENDSELECT

;ABL:마조끼를 본다
SELECTCASE ABL:ARG:마조끼
	CASE 0
		TIMES LOCAL , 1.00
	CASE 1 TO 3
		TIMES LOCAL , 0.90
	CASE 3 TO 5
		TIMES LOCAL , 0.75
	CASE 6,7
		TIMES LOCAL , 0.60
	CASE 8,9
		TIMES LOCAL , 0.50
	CASE 10 TO 12
		TIMES LOCAL , 0.30
	CASE 13 TO 15
		TIMES LOCAL , 0.20
	CASEELSE
		TIMES LOCAL , 0.10
ENDSELECT
CUP:ARG:반감 += LOCAL

license: mit


code:
@사정절정표기(ARG)
#DIM C절정회수
#DIM V절정회수
#DIM A절정회수
#DIM B절정회수
#DIM M절정회수
#DIM 동시절정보정
#DIM 개행판정
#DIM 접속판정
#DIMS 이름표기

IF ARG == MASTER
	이름표기 =
ELSE
	이름표기 = (%CALLNAME:ARG%)
ENDIF

;●---절정의 표기---●
IF NOWEX:ARG:오중절정
	동시절정보정 = 16
ELSEIF NOWEX:ARG:사중절정
	동시절정보정 = 8
ELSEIF NOWEX:ARG:삼중절정
	동시절정보정 = 4
ELSEIF NOWEX:ARG:이중절정
	동시절정보정 = 2
ELSE
	동시절정보정 = 1
ENDIF

C절정회수 = NOWEX:ARG:C절정 / 동시절정보정
V절정회수 = NOWEX:ARG:V절정 / 동시절정보정
A절정회수 = NOWEX:ARG:A절정 / 동시절정보정
B절정회수 = NOWEX:ARG:B절정 / 동시절정보정
M절정회수 = NOWEX:ARG:M절정 / 동시절정보정

SETCOLOR C_PINK
개행판정 = 0
LOCALS:0 =
LOCALS:1 =
LOCALS:2 =
IF C절정회수 > 2
	LOCALS:2 += "C"
ELSEIF C절정회수 > 1
	LOCALS:1 += "C"
ELSEIF C절정회수 > 0
	LOCALS:0 += "C"
ENDIF
IF V절정회수 > 2
	LOCALS:2 += "V"
ELSEIF V절정회수 > 1
	LOCALS:1 += "V"
ELSEIF V절정회수 > 0
	LOCALS:0 += "V"
ENDIF
IF A절정회수 > 2
	LOCALS:2 += "A"
ELSEIF A절정회수 > 1
	LOCALS:1 += "A"
ELSEIF A절정회수 > 0
	LOCALS:0 += "A"
ENDIF
IF B절정회수 > 2
	LOCALS:2 += "B"
ELSEIF B절정회수 > 1
	LOCALS:1 += "B"
ELSEIF B절정회수 > 0
	LOCALS:0 += "B"
ENDIF
IF M절정회수 > 2
	LOCALS:2 += "M"
ELSEIF M절정회수 > 1
	LOCALS:1 += "M"
ELSEIF M절정회수 > 0
	LOCALS:0 += "M"
ENDIF
IF LOCALS:0 != "" || LOCALS:1 != "" || LOCALS:2 != ""
	PRINT 《
	접속판정 = 0
	IF LOCALS:2 != ""
		PRINTFORM %LOCALS:2%초강절정
		접속판정 = 1
	ENDIF
	IF LOCALS:1 != ""
		IF 접속판정
			PRINT /
		ENDIF
		PRINTFORM %LOCALS:1%강절정
		접속판정 = 1
	ENDIF
	IF LOCALS:0 != ""
		IF 접속판정
			PRINT /
		ENDIF
		PRINTFORM %LOCALS:0%절정
	ENDIF
	PRINT 》
	개행판정 = 1
ENDIF
SIF SYNCED_ORGASM(ARG) && !FLAG:70
	CALL COLORMESSAGE(@"＜동시절정＞（호감도, 소스 입수량 증가）",C_YELLOW)

;●---사정의 표기---●
SETCOLOR C_CREAM
IF NOWEX:ARG:사정
	PRINT 【
	SELECTCASE TCVAR:ARG:12
		CASE 1
			PRINT 질내
		CASE 2
			PRINT 항내
		CASE 4
			PRINT 입안
		CASE 5
			PRINT 유내
		CASE 12
			PRINT A고문
		CASE 13
			PRINT 질고문
	ENDSELECT
	SIF NOWEX:ARG:사정 > 2
		PRINT 초과
	SIF NOWEX:ARG:사정 > 1
		PRINT 대량
	SIF TALENT:ARG:농후정액
		PRINT 농후
	PRINT 사정】
	개행판정 = 1
ENDIF

RESETCOLOR

SIF 개행판정
	PRINTFORML %이름표기%

IF ARG == MASTER
	FOR LOCAL,1,CHARANUM
		CALL ADDICTION_SHOW(LOCAL, "질내사정보정", "질내사정중독")
		CALL ADDICTION_SHOW(LOCAL, "항내사정보정", "항내사정중독")
		CALL ADDICTION_SHOW(LOCAL, "구강사정보정", "정액중독")
	NEXT
ENDIF


code:
@SAMEN_DIRECTION(ARG,ARG:1)
#FUNCTION
;ARG 사정하는 사람
;ARG:1 사정되는 사람
LOCALS = /{ARG:1}/
SIF STRCOUNT(CSTR:ARG:10,LOCALS)
	RETURNF 1

@SAMEN_DIRECTION2(ARG,ARG:1)
#FUNCTION
;ARG 사정하는 사람
;ARG:1 사정되는 사람
LOCALS = /{ARG:1}/
SIF STRCOUNT(CSTR:ARG:11,LOCALS)
	RETURNF 1

@EVENT_SHOOT
#DIM PARTICIPANTS	;참가인원수
VARSET LOCAL
;TCVAR:2 사정장소 플래그(1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
;TCVAR:4 사정쾌감강도
;사정 전판정
FOR LOCAL,0,CHARANUM
	SIF !HAS_PENIS(LOCAL)
		CONTINUE
	;콘돔
	IF TEQUIP:LOCAL:콘돔
		TCVAR:LOCAL:콘돔 = 1
	ELSE
		TCVAR:LOCAL:콘돔 = 0
	ENDIF
	SIF LOCAL > 0
		CALL TARGET_EJAC_CHECK(LOCAL)
	;더블 펠라등의 두 사람째보정에 사용한다
	PARTICIPANTS = 0
	FOR LOCAL:1,0,CHARANUM
		IF LOCAL:1 == 0
			LOCAL:3 = 0
		ELSE
			SIF TARGET:(LOCAL:1) <= 0
				CONTINUE
			LOCAL:3 = TARGET:(LOCAL:1)
		ENDIF
		SIF SAMEN_DIRECTION(LOCAL,LOCAL:3)
			CALL EJAC_CHECK(LOCAL,LOCAL:3,PARTICIPANTS)
		PARTICIPANTS ++
	NEXT
NEXT
;사정 처리
FOR LOCAL, 0, GET_TARGETNUM() + 1
	LOCAL:1 = TARGET:LOCAL
	SIF !LOCAL
		LOCAL:1 = MASTER
	;사정 처리
	SIF TCVAR:(LOCAL:1):사정쾌감강도
		CALL EJAC_CHECK_AFTER(LOCAL:1)
	SIF NOWEX:(LOCAL:1):사정
		CALL SAMEN_SHOOT(LOCAL:1)
NEXT

;필수는 아니지만 말표시를 알기 쉽게 하기 위해서 사정 상황을 TFLAG:1에 할당한다(현재 상태로서는 필수입니다)
;PLAYER로부터 사정 대상에의 질내~애널 봉사 사정까지는 TCVAR:12(=1~9)에 준거
SIF SAMEN_DIRECTION2(PLAYER,MASTER_POSE(0,SET_P))
	SETBIT TFLAG:1,TCVAR:PLAYER:12

FOR LOCAL,1,CHARANUM
	;콘돔 판정（캐릭터분）
	IF TCVAR:LOCAL:콘돔 == 1 && NOWEX:LOCAL:사정
		;20190823 현재는 없지만 캐릭터 측에 콘돔을 착용한 경우를 추가한다면 여기를 상정
	ENDIF
	IF SAMEN_DIRECTION2(PLAYER,LOCAL)
		SETBIT TCVAR:LOCAL:50,TCVAR:PLAYER:12
	ENDIF
NEXT

;사정 부분이 특수한 경우

;콘돔 판정（아나타분）
IF TCVAR:PLAYER:콘돔 == 1 && NOWEX:PLAYER:사정
	SETBIT TFLAG:1,0
	SETBIT TCVAR:50,0
ENDIF
;수음 펠라
IF TFLAG:50 == 5 && NOWEX:PLAYER:사정
	SETBIT TFLAG:1,20
	SETBIT TCVAR:50,20
ENDIF
;파이즈리 펠라
IF TFLAG:50 == 12 && NOWEX:PLAYER:사정
	SETBIT TFLAG:1,21
	SETBIT TCVAR:50,21
ENDIF
;授乳プレイ
IF TFLAG:50 == 19 && NOWEX:PLAYER:사정
	SETBIT TFLAG:1,26
	SETBIT TCVAR:50,26
ENDIF
;거품춤으로 TARGET 사정
IF TCVAR:PLAYER:12 == 8
	SETBIT TFLAG:1,8
	SETBIT TCVAR:50,8
ENDIF
;식스나인
IF TFLAG:50 == 1 && (NOWEX:PLAYER:사정 || NOWEX:사정)
	SETBIT TFLAG:1,22
	SETBIT TCVAR:50,22
ENDIF
;이라마치오
IF SELECTCOM == 140 && NOWEX:PLAYER:사정
	SETBIT TFLAG:1,23
	SETBIT TCVAR:50,23
ENDIF
;자궁구 고문
IF TFLAG:50 == 7 && NOWEX:PLAYER:사정
	SETBIT TFLAG:1,24
	SETBIT TCVAR:50,24
ENDIF
SIF !GETBIT(TFLAG:일일일회,8) && NOWEX:PLAYER:사정
	SETBIT TFLAG:일일일회,8

@EJAC_CHECK(ARG, PRIMARY, SECONDARY)
#DIM 기교보정
#DIM 게이지증가량
#DIM 총계쾌락강도
#DIM P감도보정
#DIM 순종보정
#DIM 욕망보정
#DIM 봉사정신보정
#DIM 윤활보정
#DIM PRIMARY	;사정시키는 사람
#DIM SECONDARY	;W펠라등에 있어서의 두 사람째보정
#DIM divisor
;-------------------------------------------------
;사정 게이지 체크
;-------------------------------------------------
;LOCAL:1~에 보정치를 처넣어 갑니다
;CFLAG:101 사정장소 플래그(1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=

;기교(공통)
LOCAL = ABL:PRIMARY:기교
기교보정 = 10 + GET_REVISION(LOCAL, 5 , 5)

;순종(공통)
LOCAL = 0
IF PRIMARY == MASTER
	LOCAL = 3
ELSE
	LOCAL = ABL:PRIMARY:순종
ENDIF
순종보정 = 10 + GET_REVISION(LOCAL, 3 , 5)

;봉사정신(봉사계)
LOCAL = 0
IF TCVAR:PRIMARY:108 && ARG == PLAYER
	LOCAL = ABL:PRIMARY:봉사정신
	봉사정신보정 = 10 + GET_REVISION(LOCAL, 10 , 5)
ELSE
	봉사정신보정 = 10
ENDIF

;욕망(성교계)
욕망보정 = 10
IF TCVAR:SECONDARY <= 2
	LOCAL = 0
	IF PRIMARY == MASTER
		LOCAL = 3
	ELSE
		LOCAL = ABL:PRIMARY:욕망
	ENDIF
	욕망보정 = 10 + GET_REVISION(LOCAL, 5 , 3)
ENDIF

;윤활
윤활보정 = 10
IF TCVAR:SECONDARY <= 2 || TCVAR:SECONDARY == 6
	IF PALAM:윤활 < PALAMLV:2
		윤활보정 = 2
	ELSEIF PALAM:윤활 < PALAMLV:3
		윤활보정 = 4
	ELSEIF PALAM:윤활 < PALAMLV:4
		윤활보정 = 6
	ELSEIF PALAM:윤활 < PALAMLV:7
		윤활보정 = 8
	ELSE
		윤활보정 = 10
	ENDIF
ENDIF

;플레이어의 C감각을 본다
P감도보정 = 10 + ABL:ARG:0 / 2

DEBUGPRINTFORML 보정 전쾌락강도{TCVAR:ARG:4}
DEBUGPRINTFORML 기교보정{기교보정}
DEBUGPRINTFORML 욕망보정{욕망보정}
DEBUGPRINTFORML 윤활보정{윤활보정}
DEBUGPRINTFORML 봉사정신보정{봉사정신보정}
DEBUGPRINTFORML P감도보정{P감도보정}
DEBUGPRINTFORML 부위보정{부위보정(ARG, PRIMARY, SECONDARY)}
총계쾌락강도 = TCVAR:ARG:4 * 기교보정 * 순종보정 * P감도보정 * 봉사정신보정 * 윤활보정 * 욕망보정 * 부위보정(ARG, PRIMARY, SECONDARY) / 10000000 * (SECONDARY + 1)

SIF CFLAG:ARG:우후후 != 2
	총계쾌락강도 = 총계쾌락강도 * (BASE:ARG:정력 + MAXBASE:ARG:정력) / (MAXBASE:ARG:정력 * 2)
SIF TCVAR:ARG:콘돔
	총계쾌락강도 = 총계쾌락강도 * 4 / 5
SIF TEQUIP:ARG:애태우기모드
	총계쾌락강도 = 총계쾌락강도 * 4 / 5
SIF TALENT:ARG:스침
	총계쾌락강도 = 총계쾌락강도 * 4 / 5
게이지증가량 = 총계쾌락강도

DEBUGPRINTFORML 보정 후 쾌락강도{총계쾌락강도}

SELECTCASE BASE:ARG:발기
	CASE IS < 300
		총계쾌락강도 /= 5
	CASE IS < 500
		총계쾌락강도 /= 4
	CASE IS < 1000
		총계쾌락강도 /= 3
	CASE IS < 1500
		총계쾌락강도 /= 2
	CASE 1500
		총계쾌락강도 = 총계쾌락강도 * 2 / 3
ENDSELECT
BASE:ARG:발기 = BASE:ARG:발기 + 총계쾌락강도 / 5

SELECTCASE BASE:ARG:발기
	CASE IS < 500
		게이지증가량 /= 5
	CASE IS < 1000
		게이지증가량 /= 3
	CASE IS < 1500
		게이지증가량 = 게이지증가량 / 5 * 2
	CASE 1500
		게이지증가량 /= 2
ENDSELECT

IF !TCVAR:ARG:이기초
	divisor = POWER(2, TCVAR:ARG:사정참기)
	게이지증가량 = 게이지증가량 / divisor
ENDIF

BASE:ARG:사정 += 게이지증가량

;PRINTFORMW %CALLNAME:(PRIMARY)%によるゲージ増加【{ゲージ増加量}】

TCVAR:ARG:사정게이지증가량 = 게이지증가량

@부위보정(ARG, PRIMARY, SECONDARY)
;-------------------------------------------------
;LOCAL 보정전의 쾌락강도
;ARG 사정하는 사람
;PRIMARY 사정시키는 사람
;SECONDARY W펠라등에 있어서의 두 사람째보정
;-------------------------------------------------
#FUNCTION
#DIM PRIMARY
#DIM SECONDARY
LOCAL = 10
;연모
SIF TALENT:PRIMARY:연모
	LOCAL += 2

SELECTCASE TCVAR:SECONDARY
	;질
	CASE 1
		;처녀
		SIF EX:PRIMARY:처녀상실체크
			LOCAL += 8
		SIF TALENT:PRIMARY:음호
			LOCAL += 3
		SIF !FLAG:70
			LOCAL += ((ABL:PRIMARY:허리 + ABL:PRIMARY:질 * 2) / 2 - 3)
		SIF NOWEX:PRIMARY:V절정
			LOCAL += 1
		SIF 삽입불가(PRIMARY) == -1 && ABL:PRIMARY:질 < 2
			LOCAL -= 2
		SIF 삽입불가(PRIMARY) == -2 && ABL:PRIMARY:질 < 3
			LOCAL -= 5
	;애널
	CASE 2
		SIF !EXP:PRIMARY:A성교경험
			LOCAL += 8
		SIF TALENT:PRIMARY:항문광
			LOCAL += 3
		SIF !FLAG:70
			LOCAL += ((ABL:PRIMARY:허리 + ABL:PRIMARY:애널 * 2) / 2 - 3)
		SIF NOWEX:PRIMARY:A절정
			LOCAL += 1
		SIF 삽입불가(PRIMARY) == -1 && ABL:PRIMARY:애널 < 2
			LOCAL -= 2
		SIF 삽입불가(PRIMARY) == -2 && ABL:PRIMARY:애널 < 3
			LOCAL -= 5
	;수음
	CASE 3
		SIF TALENT:PRIMARY:능숙한손가락
			LOCAL += 2
		SIF !FLAG:70 && !CFLAG:장난
			LOCAL += (ABL:PRIMARY:손가락 - 2) / 2
	;구음
	CASE 4
		SIF TALENT:PRIMARY:혀놀림
			LOCAL += 2
		SIF TALENT:PRIMARY:고양이혀
			LOCAL -= 2
		SIF !FLAG:70 && !CFLAG:PRIMARY:장난
			LOCAL += (ABL:PRIMARY:혀 - 2) / 2
		;파이즈리 펠라
		IF TFLAG:50 == 12
			LOCAL += TALENT:PRIMARY:가슴사이즈
			IF !FLAG:70 && !CFLAG:PRIMARY:장난
				SELECTCASE ABL:PRIMARY:가슴
					CASE 0 TO 1
						LOCAL = LOCAL - 6 + ABL:PRIMARY:가슴
					CASE IS >= 2
						LOCAL += (ABL:PRIMARY:가슴 - 2) / 2
				ENDSELECT
			ENDIF
		ENDIF
		;내친 김에 청소 플래그
		IF ABL:PRIMARY:봉사정신 >= 2 && ABL:PRIMARY:기교 >= 2 && TEQUIP:PRIMARY:볼개그 == 0 && EXP:PRIMARY:정음경험 > 20
			STAIN:SECONDARY = 2
			IF TFLAG:50 == 9 && PRIMARY == MASTER_POSE(SET_M,SET_P,1)
				TFLAG:8 |= 2
			ELSEIF PRIMARY == TARGET
				TFLAG:8 |= 1
			ENDIF
		ENDIF
	;파이즈리
	CASE 5
		LOCAL += TALENT:PRIMARY:가슴사이즈
		IF !FLAG:70 && !CFLAG:PRIMARY:장난
			SELECTCASE ABL:PRIMARY:가슴
				CASE 0 TO 1
					LOCAL = LOCAL - 6 + ABL:PRIMARY:가슴
				CASE IS >= 2
					LOCAL += (ABL:PRIMARY:가슴 - 2) / 2
			ENDSELECT
		ENDIF
		;파이즈리 펠라
		IF TFLAG:50 == 12
			SIF TALENT:PRIMARY:혀놀림
				ARG += 2
			SIF TALENT:PRIMARY:고양이혀
				ARG -= 2
			LOCAL += (ABL:PRIMARY:혀 - 2) / 2
		ENDIF
		;청소 플래그
		IF ABL:PRIMARY:봉사정신 >= 2 && ABL:PRIMARY:기교 >= 2 && TEQUIP:PRIMARY:볼개그 == 0 && EXP:PRIMARY:정음경험 > 20
			STAIN:SECONDARY = 2
			IF PRIMARY == ASSI
				TFLAG:8 |= 2
			ELSEIF PRIMARY == TARGET
				TFLAG:8 |= 1
			ENDIF
		ENDIF
	;가랑이
	CASE 6
	;풋잡
	CASE 7
		SIF TALENT:PRIMARY:새드
			LOCAL += 2
	;체표(거품춤)
	CASE 8
	;애널 봉사
	CASE 9
		SIF TALENT:PRIMARY:혀놀림
			LOCAL += 2
		SIF TALENT:PRIMARY:고양이혀
			LOCAL -= 2
		SIF !FLAG:70
			LOCAL += (ABL:PRIMARY:혀 - 2) / 2
	;도구
	CASE 10
	;촉수
	CASE 11
	;A고문
	CASE 12
	;V고문
	CASE 13
ENDSELECT

RETURNF LOCAL


code:
;사정 확정 후의 처리
@EJAC_CHECK_AFTER(캐릭터등록번호)
#DIM 캐릭터등록번호
#DIM 사정회수
#DIM 과잉사정

VARSET 사정회수
VARSET 과잉사정
;캐릭터등록번호 사정하는 사람
IF !TCVAR:캐릭터등록번호:이기초
	IF BASE:캐릭터등록번호:사정 > MAXBASE:캐릭터등록번호:사정
		IF BASE:캐릭터등록번호:발기 < 1000
			BASE:캐릭터등록번호:사정 = MAXBASE:캐릭터등록번호:사정 - 1
		ELSE
			TCVAR:캐릭터등록번호:이기초 = 1
		ENDIF
	ENDIF
	RETURN 0
ELSEIF TEQUIP:캐릭터등록번호:애태우기모드 && 캐릭터등록번호 != MASTER
	TCVAR:캐릭터등록번호:초조하게해회수 += 2
	RETURN 0
ENDIF

;사정(절정) 회수
사정회수 = MIN(BASE:캐릭터등록번호:사정 / MAXBASE:캐릭터등록번호:사정, (100 + BASE:캐릭터등록번호:정력) / 100)

NOWEX:캐릭터등록번호:사정량 = BASE:캐릭터등록번호:사정 * 10 / (ABL:캐릭터등록번호:C감각 / 2 + 10) / 100
NOWEX:캐릭터등록번호:사정량 = NOWEX:캐릭터등록번호:사정량 * 10 / (ABL:캐릭터등록번호:C감각 / 2 + 10)
SIF NOWEX:캐릭터등록번호:사정량 > 400
	NOWEX:캐릭터등록번호:사정량 = 300 + (NOWEX:캐릭터등록번호:사정량 - 300) * 7 / 10

IF NOWEX:캐릭터등록번호:사정량 > BASE:캐릭터등록번호:정력
	NOWEX:캐릭터등록번호:사정량 = (NOWEX:캐릭터등록번호:사정량 - BASE:캐릭터등록번호:정력) / 5 + BASE:캐릭터등록번호:정력
	과잉사정 = NOWEX:캐릭터등록번호:사정량 - BASE:캐릭터등록번호:정력
ENDIF

IF 과잉사정
	IF TALENT:캐릭터등록번호:절륜 ;절륜 관련 KR판 구잔재 또는 오리지날
	ELSEIF TFLAG:포이즌바디
		DOWNBASE:캐릭터등록번호:체력 += 과잉사정 * 10
	ELSE
		DOWNBASE:캐릭터등록번호:체력 += 과잉사정 * 4
	ENDIF
ENDIF

BASE:캐릭터등록번호:정력 = MAX(0, BASE:캐릭터등록번호:정력 - NOWEX:캐릭터등록번호:사정량)
TCVAR:캐릭터등록번호:이기초 = 0
STAIN:캐릭터등록번호:2 |= 4

;BASE의 처리
BASE:캐릭터등록번호:사정 = 0

CSTR:캐릭터등록번호:사정후처리용 = %CSTR:캐릭터등록번호:사정판정용%
TCVAR:캐릭터등록번호:사정한장소플래그 = TCVAR:캐릭터등록번호:사정장소

;이상경험, 사정경험, 정액경험
;여성기가 있는 경우?
SIF !EXP:캐릭터등록번호:사정경험 && TALENT:캐릭터등록번호:성별 & 1
	EXP:캐릭터등록번호:이상경험 += 1

EXP:캐릭터등록번호:절정경험 ++
; MAXBASE:캐릭터등록번호:정력 += 사정회수*3;;정력증가;KR판 구잔재 또는 오리지날
CALL 사정정력최대치증가(캐릭터등록번호, 사정회수 * 3) ;개인 임의수정, 사정에 의한 정력증가를 함수로 처리
EXP:캐릭터등록번호:사정경험 += 사정회수

NOWEX:캐릭터등록번호:사정 += 사정회수
EX:캐릭터등록번호:사정 += 사정회수
TCVAR:캐릭터등록번호:초조하게해회수 = 0
TEQUIP:캐릭터등록번호:애태우기모드 = 0
SIF AT_HOME(캐릭터등록번호)
	YOGORE:(CFLAG:캐릭터등록번호:현재위치) += 사정회수 * 50

IF (TALENT:캐릭터등록번호:절륜);KR판 고유 또는 오리지날. 원본은 무조건 사정량 초과시 감소
ELSEIF TFLAG:포이즌바디
	DOWNBASE:캐릭터등록번호:체력 += LIMIT(NOWEX:캐릭터등록번호:사정량 / 4, 30, 300)
	DOWNBASE:캐릭터등록번호:기력 += LIMIT(NOWEX:캐릭터등록번호:사정량 / 2, 40, 400)
ELSE
	DOWNBASE:캐릭터등록번호:체력 += LIMIT(NOWEX:캐릭터등록번호:사정량 / 8, 10, 100)
	DOWNBASE:캐릭터등록번호:기력 += LIMIT(NOWEX:캐릭터등록번호:사정량 / 4, 20, 200)
ENDIF

;MASTER가 아니면 BASE나 SOURCE등의 처리를 넣는다
SIF 캐릭터등록번호 == MASTER
	RETURN 0

IF EXP:캐릭터등록번호:사정경험 < EXPLV:1
	SOURCE:캐릭터등록번호:노출 += 5000 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:굴종 += 2500 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:불결 += 200 * (사정회수 + 1) / 2
ELSEIF EXP:캐릭터등록번호:사정경험 < EXPLV:2
	SOURCE:캐릭터등록번호:노출 += 2500 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:굴종 += 2000 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:불결 += 150 * (사정회수 + 1) / 2
ELSEIF EXP:캐릭터등록번호:사정경험 < EXPLV:3
	SOURCE:캐릭터등록번호:노출 += 1000 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:굴종 += 800 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:불결 += 100 * (사정회수 + 1) / 2
ELSEIF EXP:캐릭터등록번호:사정경험 < EXPLV:4
	SOURCE:캐릭터등록번호:노출 += 800 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:굴종 += 600 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:불결 += 50 * (사정회수 + 1) / 2
ELSEIF EXP:캐릭터등록번호:사정경험 < EXPLV:5
	SOURCE:캐릭터등록번호:노출 += 400 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:굴종 += 400 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:불결 += 10 * (사정회수 + 1) / 2
ELSE
	SOURCE:캐릭터등록번호:노출 += 200 * (사정회수 + 1) / 2
	SOURCE:캐릭터등록번호:굴종 += 250 * (사정회수 + 1) / 2
ENDIF

;-------------------------------------------------
;조교 대상의 사정 체크
;-------------------------------------------------
@TARGET_EJAC_CHECK(캐릭터등록번호)
#DIM 캐릭터등록번호

;남성기가 없으면 돌아간다
SIF !HAS_PENIS(캐릭터등록번호)
	RETURN 0

LOCAL = (CUP:캐릭터등록번호:쾌C + CUP:캐릭터등록번호:쾌V + CUP:캐릭터등록번호:쾌A + CUP:캐릭터등록번호:쾌B) / 4

;자제심
SIF TALENT:캐릭터등록번호:자제심
	LOCAL /= 2

;쾌감에 솔직
SIF TALENT:캐릭터등록번호:쾌감응답 > 0
	TIMES LOCAL , 1.20

;쾌감에 부정(음란, 창부로 무효)
SIF TALENT:캐릭터등록번호:쾌감응답 < 0 && !TALENT:캐릭터등록번호:음란
	TIMES LOCAL , 0.80

;미약
SIF (TCVAR:캐릭터등록번호:미약 || IsInsenceValid(캐릭터등록번호)) && 캐릭터등록번호 != MASTER
	LOCAL *= 2

;仙香玉兎
SIF IsInsenceValid(캐릭터등록번호)
	LOCAL *= 2

;이뇨제
SIF TCVAR:캐릭터등록번호:이뇨제
	LOCAL /= 2
;수면중
SIF CFLAG:캐릭터등록번호:수면
	LOCAL /= 2
BASE:캐릭터등록번호:사정 += LOCAL
;SIF BASE:캐릭터등록번호:사정 > MAXBASE:캐릭터등록번호:사정
;	BASE:캐릭터등록번호:사정 = MAXBASE:캐릭터등록번호:사정 - 1

;사정 처리의 캐릭터별 터미널
@SAMEN_SHOOT(사정한캐릭터번호)
#DIM 사정한캐릭터번호
#DIM 사정된캐릭터번호

FOR 사정된캐릭터번호, 0, CHARANUM
	SIF SAMEN_DIRECTION2(사정한캐릭터번호, 사정된캐릭터번호)
		CALL SAMEN_SHOOT2(사정한캐릭터번호, 사정된캐릭터번호)
NEXT

;부위별 처리
@SAMEN_SHOOT2(SENDER, RECEIVER)
#DIM SENDER
#DIM RECEIVER
#DIM 억지로반감
#DIM 억지로공포
#DIM 신뢰마이나스
#DIM 호감도마이너스
#DIM 욕구불만해소계수
#DIM 중독보너스
#DIM 모브코용

CALL TWKR_ORIG_SAMEN(RECEIVER, SENDER)

중독보너스 = 0
IF SENDER == MASTER && CFLAG:RECEIVER:우후후 == 2
	SOURCE:RECEIVER:달성 += NOWEX:SENDER:사정 * 150
	SOURCE:RECEIVER:정복 += NOWEX:SENDER:사정 * 150
	SIF !(NOWEX:RECEIVER:C절정 + NOWEX:RECEIVER:V절정 + NOWEX:RECEIVER:A절정 + NOWEX:RECEIVER:B절정 + NOWEX:RECEIVER:M절정)
		SOURCE:RECEIVER:정복 += NOWEX:SENDER:사정 * 100
ENDIF
IF SENDER == MASTER && TALENT:RECEIVER:스침 == 1
	모브코용 = MIN(NOWEX:SENDER:사정,3)
	SELECTCASE TCVAR:SENDER:사정한장소플래그
		CASE 1
			SIF ABL:RECEIVER:질 < 모브코용
				CALL ChangeABL(RECEIVER, "질", 1)
		CASE 2
			SIF ABL:RECEIVER:애널 < 모브코용
				CALL ChangeABL(RECEIVER, "애널", 1)
		CASE 3
			SIF ABL:RECEIVER:손가락 < 모브코용
				CALL ChangeABL(RECEIVER, "손가락", 1)
		CASE 4
			SIF ABL:RECEIVER:혀 < 모브코용
				CALL ChangeABL(RECEIVER, "혀", 1)
		CASE 4
			SIF ABL:RECEIVER:가슴 < 모브코용
				CALL ChangeABL(RECEIVER, "가슴", 1)
	ENDSELECT
ENDIF

IF TEQUIP:SENDER:콘돔
	;고무를 제외한다
	TEQUIP:SENDER:콘돔 = 0
	IF RECEIVER
		SELECTCASE TCVAR:SENDER:사정한장소플래그
			CASE 1
				CFLAG:RECEIVER:욕구불만도 += ABL:RECEIVER:질내사정중독 * NOWEX:SENDER:사정 * 15
			CASE 2
				CFLAG:RECEIVER:욕구불만도 += ABL:RECEIVER:항내사정중독 * NOWEX:SENDER:사정 * 15
		ENDSELECT
	ENDIF
	;접촉의 리셋
	CALL 삽입해제(RECEIVER)
	TFLAG:고무내 = NOWEX:SENDER:사정량
	IF CFLAG:RECEIVER:자동고무 && ITEM:콘돔 && SENDER == MASTER
		ITEM:콘돔 --
		TEQUIP:SENDER:콘돔 = 1
	ENDIF
ELSE
;TCVAR:12 사정장소 플래그(1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수)
SELECTCASE TCVAR:SENDER:사정한장소플래그
	;질내
	CASE 1
		EXP:RECEIVER:정액경험 += NOWEX:SENDER:사정 * 2
		EXP:RECEIVER:질내사정경험 += NOWEX:SENDER:사정 * 2
		STAIN:RECEIVER:3 |= 4
		STAIN:RECEIVER:질내 |= 4
		TCVAR:RECEIVER:V에질내사정된플래그 = 2
		TCVAR:RECEIVER:V에질내사정한캐릭터의번호 = SENDER
		IF FLAG:시간정지 == 1
			CFLAG:RECEIVER:질내사정 = 2
		ELSE
			CFLAG:RECEIVER:질내사정 = 1
		ENDIF
		NOWEX:RECEIVER:질내정액 = NOWEX:SENDER:사정량
		EX:RECEIVER:질내정액 += NOWEX:RECEIVER:질내정액
		욕구불만해소계수 = 800 / MIN(100 + ABL:RECEIVER:질내사정중독 * 10,200)
		CFLAG:RECEIVER:욕구불만도 -= SQRT(NOWEX:RECEIVER:질내정액) * 욕구불만해소계수
		CFLAG:RECEIVER:누계질내정액 += NOWEX:RECEIVER:질내정액
		CALL POISONBODY(RECEIVER, SENDER, 1)
		;SIF !TCVAR:RECEIVER:피임약
		;PRINTFORML %CALLNAME:SENDER%が%CALLNAME:RECEIVER%に{NOWEX:RECEIVER:질내정액}出した
		CALL SAMEN_SHOOT3(NOWEX:RECEIVER:질내정액 , SENDER, RECEIVER)
		IF SHIRAHU(RECEIVER)
			TCVAR:RECEIVER:씨받음자각 = 1
		ENDIF
		IF SHIRAHU(SENDER) && SENDER != MASTER
			TCVAR:SENDER:씨뿌림자각 = 1
		ENDIF
		IF TCVAR:RECEIVER:억지로
			IF TALENT:RECEIVER:스침 == 1
				CSTR:RECEIVER:모브코프로필2 = %REPLACE(@"%CSTR:RECEIVER:모브코프로필2%","당해 지금은 얌전하다","당해 질내사정되었다")%
				SIF !STRCOUNT(CSTR:RECEIVER:모브코프로필2, "질내사정되었다")
					CSTR:RECEIVER:모브코프로필2 = %REPLACE(@"%CSTR:RECEIVER:모브코프로필2%","처녀를 빼앗겼다","처녀를 빼앗긴 끝에 질내사정되었다")%
				IF EX:RECEIVER:질내정액 >= 300 && !STRCOUNT(CSTR:RECEIVER:모브코프로필2, "듬뿍 ")
					CSTR:RECEIVER:모브코프로필2 = %REPLACE(@"%CSTR:RECEIVER:모브코프로필2%","질내사정되었다","듬뿍 정액을 쏟아부어졌다")%
				ENDIF
				IF 모브코초기V성교경험 + 10 <= EXP:RECEIVER:V성교경험
					IF !STRCOUNT(CSTR:RECEIVER:모브코프로필2, "박혀 쓰러져")
						CSTR:RECEIVER:모브코프로필2 = %REPLACE(@"%CSTR:RECEIVER:모브코프로필2%","질내사정되었다", "박혀 쓰러져 질내사정되었다")%
					ENDIF
					IF 모브코초기V성교경험 + 15 <= EXP:RECEIVER:V성교경험 && !STRCOUNT(CSTR:RECEIVER:모브코프로필2, "마구")
						CSTR:RECEIVER:모브코프로필2 = %REPLACE(@"%CSTR:RECEIVER:모브코프로필2%","박혀 쓰러져", "마구 박혀 쓰러져")%
					ENDIF
				ENDIF
				IF MARK:RECEIVER:쾌락각인 && !STRCOUNT(CSTR:RECEIVER:모브코프로필2, "가버려")
					CSTR:RECEIVER:모브코프로필2 = %REPLACE(@"%CSTR:RECEIVER:모브코프로필2%","박혀 쓰러져", "박혀 쓰러져 가버려")%
				ENDIF
				IF MARK:RECEIVER:쾌락각인 && !STRCOUNT(CSTR:RECEIVER:모브코프로필2, "미친듯 가버려")
					CSTR:RECEIVER:모브코프로필2 = %REPLACE(@"%CSTR:RECEIVER:모브코프로필2%","가버려", "미친듯 가버려")%
				ENDIF
			ENDIF
			SELECTCASE ABL:RECEIVER:질내사정중독
				CASE IS < 3
					억지로반감 = 100000
					호감도마이너스 = NOWEX:SENDER:사정 * RAND:50 + NOWEX:SENDER:사정 * RAND:50
					신뢰마이나스 = NOWEX:SENDER:사정 * RAND:20 + NOWEX:SENDER:사정 * RAND:20
					억지로공포 = NOWEX:RECEIVER:질내정액 * 5
				CASE 3 TO 5
					억지로반감 = 50000
					호감도마이너스 = NOWEX:SENDER:사정 * RAND:30 + NOWEX:SENDER:사정 * RAND:30
					신뢰마이나스 = NOWEX:SENDER:사정 * RAND:15 + NOWEX:SENDER:사정 * RAND:15
					억지로공포 = NOWEX:RECEIVER:질내정액 * 3
				CASEELSE
					억지로반감 = 10000
					호감도마이너스 = NOWEX:SENDER:사정 * RAND:10 + NOWEX:SENDER:사정 * RAND:10
					신뢰마이나스 = NOWEX:SENDER:사정 * RAND:10 + NOWEX:SENDER:사정 * RAND:10
					억지로공포 = NOWEX:RECEIVER:질내정액
			ENDSELECT
			IF CFLAG:RECEIVER:생리주기 == 7
				억지로반감 *= 2
				억지로공포 *= 2
				호감도마이너스 *= 2
				신뢰마이나스 *= 2
			ENDIF
			IF CFLAG:RECEIVER:억지로질내사정 > 1
				억지로반감 *= 2
				억지로공포 *= 2
				호감도마이너스 *= 2
				신뢰마이나스 *= 2
			ENDIF
			SOURCE:RECEIVER:반감 += 억지로반감
			IF CFLAG:RECEIVER:억지로질내사정 > 1
				CFLAG:RECEIVER:억지로질내사정 = 3
			ELSE
				CFLAG:RECEIVER:억지로질내사정 = 1
			ENDIF
			TFLAG:호감도관리 -= 호감도마이너스
			CALL CHANGE_CFLAG(4,RECEIVER,신뢰마이나스 * (-1))
		ENDIF

		;비디오;KR판 구잔재
		IF TEQUIP:비디오촬영
			LOCALS = SV/
			TSTR:1 += LOCALS
		ENDIF
	;애널
	CASE 2
		EXP:RECEIVER:정액경험 += NOWEX:SENDER:사정 * 2
		EXP:RECEIVER:항내사정경험 += NOWEX:SENDER:사정 * 2
		STAIN:RECEIVER:4 |= 4
		STAIN:RECEIVER:장내 |= 4
		;질내사정 표시용
		TCVAR:RECEIVER:A에질내사정된플래그 = 2
		TCVAR:RECEIVER:A에질내사정한캐릭터의번호 = SENDER
		IF FLAG:시간정지 == 1
			CFLAG:RECEIVER:애널사정 = 2
		ELSE
			CFLAG:RECEIVER:애널사정 = 1
		ENDIF
		NOWEX:RECEIVER:애널내정액 = NOWEX:SENDER:사정량
		EX:RECEIVER:애널내정액 += NOWEX:SENDER:사정량
		욕구불만해소계수 = 800 / MIN(100 + ABL:RECEIVER:항내사정중독 * 10,200)
		CFLAG:RECEIVER:욕구불만도 -= SQRT(NOWEX:RECEIVER:애널내정액) * 욕구불만해소계수
		CFLAG:RECEIVER:누계애널정액 += NOWEX:SENDER:사정량
		CALL POISONBODY(RECEIVER, SENDER, 2)
	;수음
	CASE 3
		EXP:RECEIVER:정액경험 += 1 + NOWEX:SENDER:사정
		STAIN:RECEIVER:손 |= 4
		IF FLAG:시간정지 == 1
			CFLAG:RECEIVER:손에정액 = 2
		ELSE
			CFLAG:RECEIVER:손에정액 = 1
		ENDIF
		IF ABL:RECEIVER:정액중독 > 2 && ABL:RECEIVER:기교 > 2
			EXP:RECEIVER:정음경험 ++
			CFLAG:RECEIVER:누계정음 += NOWEX:SENDER:사정량 / 2
			CFLAG:RECEIVER:욕구불만도 -= NOWEX:SENDER:사정량 * ABL:RECEIVER:정액중독 / 10 * (TALENT:RECEIVER:정애미각 + 1)
			CALL POISONBODY(RECEIVER, SENDER, 31)
		ELSE
			CALL POISONBODY(RECEIVER, SENDER, 3)
		ENDIF
		;접촉의 리셋
		TEQUIP:SENDER:손가락접촉부위 = 0
	;구음
	CASE 4
		EXP:RECEIVER:정액경험 += NOWEX:SENDER:사정 * 3
		STAIN:RECEIVER:구 |= 4
		IF FLAG:시간정지 == 1
			CFLAG:RECEIVER:구내정액 = 2
		ELSE
			CFLAG:RECEIVER:구내정액 = 1
		ENDIF
		;이라마치오
		욕구불만해소계수 = 800 / MIN(100 + ABL:RECEIVER:정액중독 * 10,200)
		IF TALENT:RECEIVER:정애미각
			SOURCE:RECEIVER:쾌M += NOWEX:SENDER:사정 * 10
			욕구불만해소계수 += 1
		ENDIF
		IF SELECTCOM == 140
			EXP:RECEIVER:정음경험 += NOWEX:SENDER:사정 * 3
			CFLAG:RECEIVER:누계정음 += NOWEX:SENDER:사정량
		ELSEIF  RAND:(MIN(60, EXP:RECEIVER:정음경험 + 1)) > 10 + RAND:30
			EXP:RECEIVER:정음경험 += NOWEX:SENDER:사정 * 3
			CFLAG:RECEIVER:누계정음 += NOWEX:SENDER:사정량
		ELSE
			EXP:RECEIVER:정음경험 += NOWEX:SENDER:사정
			CFLAG:RECEIVER:누계정음 += NOWEX:SENDER:사정량 / 3
			욕구불만해소계수 /= 2
		ENDIF
		CALL POISONBODY(RECEIVER, SENDER, 4)
		CFLAG:RECEIVER:욕구불만도 -= SQRT(NOWEX:SENDER:사정량) * 욕구불만해소계수 / 4

		;접촉의 리셋
		TEQUIP:SENDER:입접촉부위 = 0
	;파이즈리
	CASE 5
		EXP:RECEIVER:정액경험 += NOWEX:SENDER:사정 * 3
		CFLAG:RECEIVER:누계뿌려짐 += NOWEX:SENDER:사정량
		CALL POISONBODY(RECEIVER, SENDER, 5)
		STAIN:RECEIVER:5 |= 4
		IF FLAG:시간정지 == 1
			CFLAG:RECEIVER:안면정액 = 2
		ELSE
			CFLAG:RECEIVER:안면정액 = 1
		ENDIF
		;TFLAG:특수COM = 12 에 해당하는 것이 CSV에 쓰여있지 않다
		;(1=식스나인 2=이와시미즈 3=G 스폿 자극 4=흐트러지고 모란 5=수음 펠라 6= 7= 8= 식스나인 파이즈리 13=교호 삽입 )
		;TFLAGS.txt라고 써 있는 것이 다르다(TFLAG:12 역강간계속 플래그(VA공통))
		;	→싫은 예감
		SIF TFLAG:특수COM == 12
			STAIN:RECEIVER:구 |= 4
		;접촉의 리셋
		TEQUIP:SENDER:B접촉부위 = 0
	;스마타
	CASE 6
		EXP:RECEIVER:정액경험 += NOWEX:SENDER:사정
		CFLAG:RECEIVER:누계뿌려짐 += NOWEX:SENDER:사정량
		CALL POISONBODY(RECEIVER, SENDER, 6)
		STAIN:RECEIVER:3 |= 4
		;접촉의 리셋
		TEQUIP:SENDER:C접촉부위 = 0
	;풋잡
	CASE 7
		EXP:RECEIVER:정액경험 += NOWEX:SENDER:사정
		CFLAG:RECEIVER:누계뿌려짐 += NOWEX:SENDER:사정량
		CALL POISONBODY(RECEIVER, SENDER, 7)
	;체표
	CASE 8
		EXP:RECEIVER:정액경험 += NOWEX:SENDER:사정 * 2
		CFLAG:RECEIVER:누계뿌려짐 += NOWEX:SENDER:사정량
		CALL POISONBODY(RECEIVER, SENDER, 8)
ENDSELECT
SIF SHIRAHU(RECEIVER)
	CALL ADDICTION_SOURCE(RECEIVER, SENDER)
ENDIF
CFLAG:RECEIVER:욕구불만도 = LIMIT(CFLAG:RECEIVER:욕구불만도, 0, 1000)
CFLAG:SENDER:욕구불만도 = MAX(CFLAG:SENDER:욕구불만도 - NOWEX:SENDER:사정 * 50, 0)

;사정된 정자를 자궁내에 보내는 처리
@SAMEN_SHOOT3(사정한정액량, 사정한캐릭터, 사정된캐릭터)
#DIM 사정한정액량
#DIM 사정된캐릭터
#DIM 감산보정치
#DIM 가산치

;사용되지 않았다……
#DIM 사정한캐릭터

;73, 삽입자궁입구고문
;뒤로부터 덧쓰기되고 있지만 이것 의미 있어?
IF SELECTCOM == 73
	감산보정치 = 100
ELSE
	SELECTCASE TEQUIP:사정된캐릭터:자궁
	CASE 0
		감산보정치 = 150
	CASE 1
		감산보정치 = 120
	CASE 2
		감산보정치 = 100
ENDSELECT
ENDIF

;정말좋아 hold
SIF TCVAR:사정된캐릭터:카운터행동의파생 == 2 && TCVAR:사정된캐릭터:카운터행동 == 500
	감산보정치 -= 30

가산치 = 사정한정액량 - 감산보정치

IF !TALENT:사정된캐릭터:스침
	SELECTCASE EXP:사정된캐릭터:자궁구개발경험
		CASE IS >= 300
		CASE IS > 100
			가산치 = 가산치 * 2 / 3
		CASEELSE
			가산치 /= 2
	ENDSELECT
ENDIF

;계산 결과가 부라면 가산하지 않는다
IF 가산치 < 0
	RETURN 0
ENDIF
SIF 사정된캐릭터 == MASTER
	TCVAR:사정한캐릭터:씨받이양 += 가산치
EX:사정된캐릭터:자궁내정액 += 가산치

;통상의 커맨드 밖에서 당신이 V에 냈을 때의 처리를 일괄로 하는 함수
;기본적으로 오나바렉스용
@ADD_SAMEN_V(ARG, 사정한정액량, 무자각)
#DIM 사정한정액량
#DIM 무자각

사정한정액량 = MIN(BASE:MASTER:정력, 사정한정액량)
BASE:MASTER:정력 -= 사정한정액량
CALL SAMEN_SHOOT3(사정한정액량, MASTER, ARG)
SIF (TALENT:MASTER:절륜);KR판 오리지날 또는 구잔재
	사정한정액량=사정한정액량*2
EX:ARG:질내정액 += 사정한정액량
CFLAG:ARG:누계질내정액 += 사정한정액량

SIF !무자각
	TCVAR:ARG:씨받음자각 = 1

STAIN:ARG:3 |= 4
STAIN:ARG:질내 |= 4

STAIN:MASTER:2 |= 1
STAIN:MASTER:2 |= 4

;통상의 커맨드 밖에서 당신이 A에 냈을 때의 처리를 일괄로 하는 함수
;기본적으로 오나바렉스용
@ADD_SAMEN_A(ARG, 사정한정액량)
#DIM 사정한정액량

사정한정액량 = MIN(BASE:MASTER:정력, 사정한정액량)

BASE:MASTER:정력 -= 사정한정액량
SIF (TALENT:MASTER:절륜);KR판 구잔재 또는 오리지날
	사정한정액량=사정한정액량*2
EX:ARG:애널내정액 += 사정한정액량
CFLAG:ARG:누계애널정액 += 사정한정액량

STAIN:ARG:4 |= 4
STAIN:ARG:장내 |= 4

STAIN:MASTER:2 |= 8
STAIN:MASTER:2 |= 4

@ADDICTION_SOURCE(RECEIVER, SENDER)
#DIM RECEIVER
#DIM SENDER
#DIM AMOUNT
#DIMS REVISED_FLAG
#DIM REVISION
#DIM 중독레벨

SELECTCASE TCVAR:SENDER:사정한장소플래그
	CASE 1
		중독레벨 = ABL:RECEIVER:질내사정중독
		REVISED_FLAG = 질내사정보정
	CASE 2
		중독레벨 = ABL:RECEIVER:항내사정중독
		REVISED_FLAG = 항내사정보정
	CASE 4
		중독레벨 = ABL:RECEIVER:정액중독
		REVISED_FLAG = 구강사정보정
	CASEELSE
		RETURN
ENDSELECT

TCVAR:RECEIVER:REVISED_FLAG = NOWEX:SENDER:사정량 * (10 + 중독레벨)

IF TALENT:SENDER:농후정액
	REVISION = GET_REVISION(TCVAR:RECEIVER:REVISED_FLAG, 500, 1000)
ELSE
	REVISION = GET_REVISION(TCVAR:RECEIVER:REVISED_FLAG, 500, 2000)
ENDIF

;동시절정의 경우 증가
SIF SENDER == MASTER && SYNCED_ORGASM(RECEIVER)
	REVISION = REVISION * 5 / 4

SOURCE:RECEIVER:사랑 += REVISION * 2
SOURCE:RECEIVER:욕정 += REVISION * 2
SOURCE:RECEIVER:공순 += REVISION
SOURCE:RECEIVER:굴종 += REVISION

TCVAR:RECEIVER:REVISED_FLAG = REVISION

@ADDICTION_SHOW(ARG, 보정명, 대응중독)
#DIMS 대응중독
#DIMS 보정명
#DIM REVISION

LOCAL = FINDELEMENT(ABLNAME, 대응중독)

REVISION = TCVAR:ARG:보정명
SIF REVISION
	PRINTFORML %CALLNAME:ARG%（%ABLNAME:LOCAL%{ABL:ARG:대응중독}）:사랑+{REVISION * 2} 욕정+{REVISION * 2} 굴종+{REVISION} 공순+{REVISION}
IF TALENT:ARG:스침 == 1 && ABL:ARG:대응중독 < REVISION / 150 && ABL:ARG:대응중독 < 3
	ABL:ARG:대응중독 ++
	CALL COLORMESSAGE(@"%CALLNAME:ARG%의 %조사처리(ABLNAME:LOCAL,"가")% {ABL:ARG:대응중독}이 되었다",C_YELLOW,2)
ENDIF

@AUTO_SKIN_EQUIP
SIF !CFLAG:TARGET:자동고무
	RETURN
SIF !ITEM:콘돔
	RETURN
SIF TEQUIP:MASTER:콘돔
	RETURN

ITEM:콘돔 --
TEQUIP:MASTER:콘돔 = 1
PRINTFORML %마스터는% 콘돔을 착용했다. 나머지 {ITEM:콘돔}매

;===========================================
;포이즌바디の効果を規定 MASTER限定
;POINTは射精箇所
;===========================================
@POISONBODY(RECEIVER, SENDER, POINT)
#DIM RECEIVER
#DIM SENDER
#DIM POINT
#DIM 게이지증가

게이지증가 = 0
IF TFLAG:포이즌바디 && SENDER == MASTER
	SELECTCASE POINT
		;膣内。粘膜吸収1.5倍。一番効果が出る
		CASE 1
			게이지증가 += NOWEX:SENDER:사정량 * 3 / 2
		;アナル。粘膜吸収1.0倍。
		CASE 2
			게이지증가 += NOWEX:SENDER:사정량
		;手淫。経皮0.2倍。一応効果はある程度
		CASE 3
			게이지증가 += NOWEX:SENDER:사정량 / 5
		;手淫→舐め取り。粘膜0.5倍
		CASE 31
			게이지증가 += NOWEX:SENDER:사정량 / 2
		;口。粘膜吸収1.0倍。飲みこぼしなどで減免はしない
		CASE 4
			게이지증가 += NOWEX:SENDER:사정량
		;胸。経皮0.5倍。手よりは多い
		CASE 5
			게이지증가 += NOWEX:SENDER:사정량 / 2
		;素股。粘膜？0.75倍
		CASE 6
			게이지증가 += NOWEX:SENDER:사정량 * 3 / 4
		;足。経皮0.2倍。一応効果はある程度
		CASE 7
			게이지증가 += NOWEX:SENDER:사정량 / 5
		;体表。経皮0.1倍。
		CASE 8
			게이지증가 += NOWEX:SENDER:사정량 / 10
	ENDSELECT
	;농후정액の場合は２倍に補正
	SIF TALENT:SENDER:농후정액
		게이지증가 *= 2
ENDIF
TCVAR:RECEIVER:포이즌게이지 += 게이지증가

;===========================================
;KR판 오리지날 기능 중 길이가 긴 부분을 따로 함수화
;===========================================
@TWKR_ORIG_SAMEN(RECEIVER, SENDER)
#DIM RECEIVER
#DIM SENDER
#DIM AblBonus
#DIM DecrementAmount
#DIM AbsorbCouple = -1, -1
#DIM DYNAMIC OriginalShoot	;개인 임의수정, 정력의 감소량을 사정량과 일치하게

; 기술에 따른 사정량 배율
; 하단에도 같은 케이스문이 있는데, 개발 편의를 위해 걍 여기서 한번 더 돌리는거야!
; 방중술보다 먼저 처리되게 설계되어있어. 생명력(?) 소진 알고리즘이 방중술로 인해서 고장나는 버그가 있어서야!
; By Subroutineuser
; AblBonus = BASE:SENDER:정력 / 50
AblBonus = MAXBASE:SENDER:정력 / 100 ;BASE 대신 MAXBASE로 처리하도록, 대신 배율감소. 3000 cap 기준 30%증가
DEBUGPRINTFORML "-SAMEN_SHOOT2.AblBonus: " + {AblBonus} + " (BEFORE HANDLE)"
; 만약 사정자가 절륜 소질을 가지고 있다면 25% 추가
IF TALENT:SENDER:절륜
	AblBonus += 25
ENDIF
;1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수
SELECTCASE TCVAR:SENDER:사정한장소플래그
	CASE 1
		AblBonus += ((ABL:SENDER:허리 + ABL:RECEIVER:질) * 2) + ((ABL:RECEIVER:기교 + ABL:SENDER:기교) / 2)
	CASE 2
		AblBonus += ((ABL:SENDER:허리 + ABL:RECEIVER:애널) * 2) + ((ABL:RECEIVER:기교 + ABL:SENDER:기교) / 2)
	CASE 3
		AblBonus += ((ABL:RECEIVER:손가락) + (ABL:RECEIVER:기교)) * 2
	CASE 4
		AblBonus += (ABL:RECEIVER:혀 * 2) + ABL:RECEIVER:기교
	CASE 5
		AblBonus += ABL:RECEIVER:기교
	CASE 6
		AblBonus += (ABL:RECEIVER:허리 * 2) + ABL:RECEIVER:기교
	CASE 7
		AblBonus += ABL:RECEIVER:기교
ENDSELECT
AblBonus = MAX(20, AblBonus)
OriginalShoot = NOWEX:SENDER:사정량
NOWEX:SENDER:사정량 += (NOWEX:SENDER:사정량 * AblBonus) / 100
IF BASE:SENDER:정력 == 0
	SETCOLOR 150, 50, 50
	; 네 정력이 아니라 생명을 짜내고 있는데?
	IF BASE:SENDER:기력 > 0
		DecrementAmount = MAX(0, (NOWEX:SENDER:사정량 * 125) / 100)
		BASE:SENDER:기력 -= DecrementAmount
		PRINTFORML 이미 한계에 도달했다...! (기력 -{DecrementAmount})
	ELSE
		DecrementAmount = NOWEX:SENDER:사정량
		BASE:SENDER:체력 -= DecrementAmount
		PRINTFORML 더 이상 견딜 수가 없다...! (체력 -{NOWEX:SENDER:사정량})
	ENDIF
	RESETCOLOR
ENDIF
; BASE:SENDER:정력 = MAX(0, BASE:SENDER:정력 - NOWEX:SENDER:사정량)
BASE:SENDER:정력 = MAX(0, BASE:SENDER:정력 - (NOWEX:SENDER:사정량 - OriginalShoot))
DEBUGPRINTFORML "-SAMEN_SHOOT2.AblBonus: " + {AblBonus} + " (AFTER HANDLE)"
; End
;;방중술 및 채양/채음 처리
IF TALENT:SENDER:방중술
	MAXBASE:SENDER:기력++
	MAXBASE:RECEIVER:기력++
	BASE:SENDER:기력 = LIMIT(BASE:SENDER:기력 + 100, 0, MAXBASE:SENDER:기력)
	BASE:RECEIVER:기력 = LIMIT(BASE:RECEIVER:기력 + 100, 0, MAXBASE:RECEIVER:기력)
ENDIF
IF GETBIT(TALENT:SENDER:179, 0) || GETBIT(TALENT:SENDER:179, 1)
	IF GETBIT(TALENT:SENDER:179,0)&&GETBIT(TALENT:SENDER:179,2) ;채음보양
		AbsorbCouple:0 = SENDER, RECEIVER
	ELSEIF GETBIT(TALENT:SENDER:179,1)&&GETBIT(TALENT:SENDER:179,3) ;채양보음
		AbsorbCouple:0 = RECEIVER, SENDER
	ELSE
		THROW 잘못된 채음보양/채양보음 값 {TALENT:SENDER:179}
	ENDIF
	MAXBASE:(AbsorbCouple:0):체력 += 2
	BASE:(AbsorbCouple:0):체력 = LIMIT(BASE:(AbsorbCouple:0):체력 + 100, 0, MAXBASE:(AbsorbCouple:0):체력)
	BASE:(AbsorbCouple:1):체력 = LIMIT(BASE:(AbsorbCouple:1):체력 - 20, 0, MAXBASE:(AbsorbCouple:1):체력)
ENDIF

;===========================================
;개인 임의수정, KR판 고유요소 사정에 의한 정력증가를 함수로 처리
;===========================================
@사정정력최대치증가(C_ID, up_value)
#DIM C_ID		;캐릭터ID
#DIM up_value 	;상승시킬값

#DIM cap_value	;상한치, 현상 3000
#DIM 실제정력

VARSET cap_value, 3000
실제정력 = MAXBASE:C_ID:정력 - BUFF:C_ID:6

IF 실제정력 < cap_value
	SIF (실제정력 + up_value) > cap_value
		up_value = cap_value - 실제정력
	MAXBASE:C_ID:정력 = MAXBASE:C_ID:정력 + up_value
ENDIF


code:
@STAIN_질내P삽입
IF !TEQUIP:PLAYER:콘돔
	;질내에 파과의 피의 더러움을 부가
	SIF TALENT:처녀 == 1 || TALENT:처녀 == 2
		STAIN:6 |= 64
	;노예의 질내⇔조교자의 P의 더러움이 이동
	STAIN:PLAYER:2 |= STAIN:3
	STAIN:PLAYER:2 |= STAIN:6
	STAIN:3 |= STAIN:PLAYER:2
	STAIN:6 |= STAIN:PLAYER:2
ENDIF

@STAIN_애널P삽입
IF !TEQUIP:PLAYER:콘돔
	;노예의 A⇔조교자의 P의 더러움이 이동
	STAIN:PLAYER:2 |= STAIN:4
	STAIN:PLAYER:2 |= STAIN:7
	STAIN:4 |= STAIN:PLAYER:2
	STAIN:7 |= STAIN:PLAYER:2
ENDIF


code:
@SOURCE_ABLUP(ARG)
IF ABL:ARG:전투능력 < 7
	IF EXP:ARG:전투경험 > EXPLV:(ABL:ARG:전투능력 + 3) || (EXP:ARG:전투경험 > 5 && ABL:ARG:전투능력 == 0)
		ABL:ARG:전투능력 ++
		PRINTFORMW %CALLNAME:ARG%의 전투능력이 LV{ABL:ARG:전투능력}이 되었습니다
		;技能レベルアップ口上,1,戦闘能力
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 1)
	ENDIF
ENDIF
IF ABL:ARG:화술기능 < 7
	IF EXP:ARG:회화경험 > EXPLV:(ABL:ARG:화술기능 + 3)
		ABL:ARG:화술기능 ++
		PRINTFORMW %CALLNAME:ARG%의 화술기능이 LV{ABL:ARG:화술기능}이 되었습니다
		;技能レベルアップ口上,2,話術技能
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 2)
	ENDIF
ENDIF

IF ABL:ARG:청소기능 < 7
	IF EXP:ARG:청소경험 > EXPLV:(ABL:ARG:청소기능 + 3)
		ABL:ARG:청소기능 ++
		PRINTFORMW %CALLNAME:ARG%의 청소기능이 LV{ABL:ARG:청소기능}이 되었습니다
		;技能レベルアップ口上,3,清掃技能
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 3)
	ENDIF
ENDIF
IF ABL:ARG:교양 < 7
	IF EXP:ARG:학습경험 > EXPLV:(ABL:ARG:교양 + 3) || (EXP:ARG:학습경험 > 5 && ABL:ARG:교양 == 0)
		ABL:ARG:교양 ++
		PRINTFORMW %CALLNAME:ARG%의 교양이 LV{ABL:ARG:교양}이 되었습니다
		;技能レベルアップ口上,4,教養
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 4)
		SIF ABL:ARG:교양 == 1 && !ARG
			PRINTW 미약을 사용할 수 있게 되었습니다
	ENDIF
ENDIF
IF ABL:ARG:요리기능 < 7
	IF EXP:ARG:요리경험 > EXPLV:(ABL:ARG:요리기능 + 3) || (EXP:ARG:요리경험 > 5 && ABL:ARG:요리기능 == 0)
		ABL:ARG:요리기능 ++
		PRINTFORMW %CALLNAME:ARG%의 요리기능이 LV{ABL:ARG:요리기능}이 되었습니다
		;技能レベルアップ口上,5,料理技能
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 5)
		IF ARG == MASTER && ABL:ARG:요리기능 >= 3 && !TALENT:MASTER:대식가
			PRINTFORMW %마스터는% 【대식가】를 얻었다
			TALENT:MASTER:대식가 = 1
		ENDIF
	ENDIF
ENDIF
IF TALENT:ARG:채집Lv < 6
	IF EXP:ARG:채집경험 > EXPLV:(TALENT:ARG:채집Lv + 3) || (EXP:ARG:채집경험 > 5 && TALENT:ARG:채집Lv == 0)
		TALENT:ARG:채집Lv ++
		PRINTFORMW %CALLNAME:ARG%의 채집Lv이 LV{TALENT:ARG:채집Lv}이 되었습니다
		;技能レベルアップ口上,6,採集Lv
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 6)
	ENDIF
ENDIF

IF ABL:ARG:음악기능 < 7
	IF (MAX(EXP:ARG:연주경험, EXP:ARG:가창경험, EXP:ARG:무도경험) > EXPLV:(ABL:ARG:음악기능 + 3) * 2) || (MAX(EXP:ARG:연주경험, EXP:ARG:가창경험, EXP:ARG:무도경험) > 20 && ABL:ARG:음악기능 == 0)
		ABL:ARG:음악기능 ++
		PRINTFORMW %CALLNAME:ARG%의 음악기능이 LV{ABL:ARG:음악기능}이 되었습니다
		SIF ARG == MASTER && ABL:ARG:음악기능 == 2
			PRINTW 바이올린을 사용할 수 있게 되었습니다
		;技能レベルアップ口上,7-0,音楽技能（7-1～はEVENTTURNENDで）
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 7, 0)

		[SKIPSTART]
		IF ABL:ARG:음악기능 != 2
		ELSEIF ARG == MASTER
			PRINTW 바이올린을 사용할 수 있게 되었습니다
		ELSEIF TALENT:ARG:음악지식
		ELSE
			;아마, 현상으로 경험을 쌓는 수단이 없다
			SELECTCASE MAX(EXP:ARG:연주경험, EXP:ARG:가창경험, EXP:ARG:무도경험)
			CASE EXP:ARG:연주경험
				PRINTFORMW 경험을 쌓은 %ARG는% 사람들 앞에서 악기의 연주를 할 수 있게 되었습니다
				TALENT:ARG:음악지식 = 1
			CASE EXP:ARG:가창경험
				PRINTFORMW 경험을 쌓은 %ARG는% 사람들 앞에서 노래를 공개 할 수가 있게 되었습니다
				TALENT:ARG:음악지식 = 2
			CASE EXP:ARG:무도경험
				PRINTFORMW 경험을 쌓은 %ARG는% 사람들 앞에서 춤을 공개 할 수가 있게 되었습니다
				TALENT:ARG:음악지식 = 3
			CASEELSE
		ENDIF
		[SKIPEND]
	ENDIF
ENDIF
IF ARG == MASTER && EXP:ARG:연주경험 >= 1000 && ABL:ARG:기교 >= 3 && !TALENT:ARG:음악지식 == 1
	PRINTFORMW %마스터는% 악기 연주에 조예가 깊어졌다
	PRINTFORMW %마스터는% [악기지식]을 얻었다
	TALENT:ARG:음악지식 = 1
ENDIF
IF ARG == MASTER && EXP:ARG:연주경험 >= 5000 && !TALENT:ARG:음감 == 1
	PRINTFORMW %CALLNAME:MASTER%의 음감은 꽤 단련할 수 있어 온 것 같다
	PRINTFORMW %마스터는% [상대 음감]을 얻었다
	TALENT:ARG:음감 = 1
ENDIF

IF TALENT:ARG:낚시Lv < 7
	IF EXP:ARG:낚시경험 > EXPLV:(TALENT:ARG:낚시Lv + 3) || (EXP:ARG:낚시경험 > 5 && TALENT:ARG:낚시Lv == 0)
		TALENT:ARG:낚시Lv ++
		PRINTFORMW %CALLNAME:ARG%의 낚시Lv이 LV{TALENT:ARG:낚시Lv}이 되었습니다
		;技能レベルアップ口上,8,釣りLv
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 8)
	ENDIF
ENDIF

IF TALENT:ARG:조제Lv < 6
	IF EXP:ARG:조제경험 > EXPLV:(TALENT:ARG:조제Lv + 3) || (EXP:ARG:조제경험 > 5 && TALENT:ARG:조제Lv == 0)
		TALENT:ARG:조제Lv ++
		PRINTFORMW %CALLNAME:ARG%의 조제Lv이 LV{TALENT:ARG:조제Lv}이 되었습니다
		;技能レベルアップ口上,9-0,調合Lv
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 9, 0)
	ENDIF
ENDIF

IF TALENT:ARG:벌채Lv < 6
	IF EXP:ARG:벌채경험 > EXPLV:(TALENT:ARG:벌채Lv + 3) || (EXP:ARG:벌채경험 > 5 && TALENT:ARG:벌채Lv == 0)
		TALENT:ARG:벌채Lv ++
		PRINTFORMW %CALLNAME:ARG%의 벌체Lv이 {TALENT:ARG:벌채Lv}이 되었습니다
		;技能レベルアップ口上,10,벌채Lv
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 10)
	ENDIF
ENDIF

IF ABL:ARG:손가락 < 7
	LOCAL = EXP:ARG:수음경험
	LOCAL:1 = 10 + ABL:ARG:기교 / 2 + TALENT:ARG:능숙한손가락 * 2 + TALENT:ARG:습득속도
	IF LOCAL * LOCAL:1 / 10 > EXPLV:(ABL:ARG:손가락 + 3) / 2; && EXP_UP(24, ARG)
		ABL:ARG:손가락 ++
		PRINTFORMW %CALLNAME:ARG%의 손가락이 능숙하게 되었다
		;技能レベルアップ口上,11-0,指
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 11, 0)
		IF ABL:ARG:손가락 >= 5 && !TALENT:ARG:능숙한손가락
			PRINTFORMW %ARG는% [능숙한손가락]을 얻었다
			TALENT:ARG:능숙한손가락 = 1
			;技能レベルアップ口上,11-1,器用な指
			CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 11, 1)
		ENDIF
	ENDIF
ENDIF
IF ABL:ARG:혀 < 7
	LOCAL = EXP:ARG:펠라티오경험
	LOCAL:1 = 10 + ABL:ARG:기교 / 2 + TALENT:ARG:혀놀림 * 2 + TALENT:ARG:습득속도
	IF  LOCAL * LOCAL:1 / 10 > EXPLV:(ABL:ARG:혀 + 3); && EXP_UP(25, ARG)
		ABL:ARG:혀 ++
		PRINTFORMW %CALLNAME:ARG%의 혀가 능숙하게 되었다
		;技能レベルアップ口上,12-0,舌
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 12, 0)
		IF ABL:ARG:혀 >= 5 && !TALENT:ARG:혀놀림
			PRINTFORMW %ARG는% [혀놀림]을 얻었다
			TALENT:ARG:혀놀림 = 1
			;技能レベルアップ口上,12-1,舌使い
			CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 12, 1)
		ENDIF
	ENDIF
ENDIF
IF ABL:ARG:가슴 < 7
	LOCAL = EXP:ARG:파이즈리경험 * 2 +  EXP:ARG:B경험 / 5 + EXP:ARG:분유경험  / 4
	LOCAL:1 = 10 + ABL:ARG:기교 / 2 + TALENT:ARG:가슴사이즈 + TALENT:ARG:습득속도
	SIF FLAG:포부 == 5
		LOCAL:1 ++
	SIF GETBIT (CFLAG:ARG:인계, 5)
		LOCAL:1 *= 4
	IF LOCAL * LOCAL:1 / 10 > EXPLV:(ABL:ARG:가슴 + 3) && EXP:ARG:파이즈리경험 + EXP:ARG:분유경험  / 5 > EXPLV:(ABL:ARG:가슴 + 1); && EXP_UP(26, ARG)
		ABL:ARG:가슴 ++
		PRINTFORMW %CALLNAME:ARG%의 가슴을 사용한 성 기술의 기술이 늘어난 것 같다
		;技能レベルアップ口上,13-0,胸
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 13, 0)
		IF ABL:ARG:가슴 >= 5 && TALENT:ARG:2 != 2 && !TALENT:ARG:음유
			PRINTFORMW %ARG는% [음유]를 얻었다
			TALENT:ARG:음유 = 1
			;技能レベルアップ口上,13-1,淫乳
			CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 13, 1)
		ENDIF
	ENDIF
ENDIF
IF ABL:ARG:허리 < 7
	LOCAL = EXP:ARG:V성교경험 + EXP:ARG:A성교경험 + EXP:ARG:삽입경험
	LOCAL:1 = 10 + ABL:ARG:기교 / 2 + TALENT:ARG:습득속도
	SIF GETBIT (CFLAG:ARG:인계, 3) || GETBIT (CFLAG:ARG:인계, 4)
		LOCAL:1 *= 4
	IF LOCAL * LOCAL:1 / 10 > EXPLV:(ABL:ARG:허리 + 3)
		ABL:ARG:허리 ++
		PRINTFORMW %CALLNAME:ARG%의 허리놀림이 향상되었다
		;技能レベルアップ口上,14,腰
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 14)
	ENDIF
ENDIF
IF ABL:ARG:질 < 7
	;너무 넓히면(자) 헐렁헐렁이 되어 버려요
	LOCAL = EXP:ARG:V성교경험  * 2 +  EXP:ARG:V경험 / 5 + (EXP:ARG:V확장경험 <= 10 ?  EXP:ARG:V확장경험 * 5 # 100 - EXP:ARG:V확장경험 * 5)
	LOCAL:1 = 10 + ABL:ARG:기교 / 2 + TALENT:ARG:습득속도
	SIF FLAG:포부 == 3
		LOCAL:1 ++
	SIF GETBIT (CFLAG:ARG:인계, 3)
		LOCAL:1 *= 4
	IF LOCAL * LOCAL:1 / 10 > EXPLV:(ABL:ARG:질 + 3) && EXP:ARG:V성교경험 >  EXPLV:(ABL:ARG:질 + 1); && EXP_UP(20, ARG)
		ABL:ARG:질 ++
		PRINTFORMW %CALLNAME:ARG%의 성교 기술이 늘어난 것 같다
		;技能レベルアップ口上,15-0,膣
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 15, 0)
		IF ABL:ARG:질 >= 5 && TALENT:ARG:2 != 2 && !TALENT:ARG:음호
			PRINTFORMW %ARG는% [음호]를 얻었다
			TALENT:ARG:음호 = 1
			;技能レベルアップ口上,15-1,淫壺
			CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 15, 1)
		ENDIF
	ENDIF
ENDIF
IF ABL:ARG:애널 < 7
	LOCAL = EXP:ARG:A성교경험  * 2 +  EXP:ARG:A경험 / 5 + (EXP:ARG:A확장경험 <= 25 ?  EXP:ARG:A확장경험 * 2 # 100 - 2 * EXP:ARG:A확장경험)
	LOCAL:1 = 10 + ABL:ARG:기교 / 2 + TALENT:ARG:습득속도
	SIF FLAG:포부 == 4
		LOCAL:1 ++
	SIF GETBIT (CFLAG:ARG:인계, 4)
		LOCAL:1 *= 4
	IF LOCAL * LOCAL:1 / 10 > EXPLV:(ABL:ARG:애널 + 3) && EXP:ARG:A성교경험 > EXPLV:(ABL:ARG:애널 + 1); && EXP_UP(21, ARG)
		ABL:ARG:애널 ++
		PRINTFORMW %CALLNAME:ARG%의 애널을 이용한 성교 기술이 늘어난 것 같다
		;技能レベルアップ口上,16,アナル
		CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 16, 0)
		IF ABL:ARG:애널 >= 5 && !TALENT:ARG:항문광
			PRINTFORMW %ARG는% [항문광]을 얻었다
			TALENT:ARG:항문광 = 1
			;技能レベルアップ口上,16,尻穴狂い
			CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 16, 1)
		ENDIF
	ENDIF
ENDIF

IF !TALENT:ARG:조합지식 && TALENT:ARG:조제Lv >= 3 && ABL:ARG:교양 >= 3
	PRINTFORML %ARG는%【조합지식】을 얻었다
	TALENT:ARG:조합지식 = 1
	;技能レベルアップ口上,9-1,調合知識
	CALL KOJO_MESSAGE_SEND("EVENT", 34, ARG, 9, 1)
ENDIF


code:
;-------------------------------------------------
;바이브 등 붙이고 있을 뿐 아이템의 체크
;-------------------------------------------------
@EQUIP_CHECK(ARG)
;클리캡 장착중
SIF TEQUIP:ARG:클리캡
	CALLFORM EQUIP_COM{COM클리캡}(ARG)

;오나홀 장착중
SIF TEQUIP:ARG:오나홀
	CALLFORM EQUIP_COM{COM오나홀}(ARG)

;바이브 장착중
SIF TEQUIP:ARG:바이브
	CALLFORM EQUIP_COM{COM바이브}(ARG)

;애널바이브 장착중
SIF TEQUIP:ARG:애널바이브
	CALLFORM EQUIP_COM{COM애널바이브}(ARG)

;애널비즈 사용중
SIF TEQUIP:ARG:애널비즈
	CALLFORM EQUIP_COM{COM애널비즈}(ARG)

;유두캡 장착중
SIF TEQUIP:ARG:유두캡
	CALLFORM EQUIP_COM{COM유두캡}(ARG)

;착유기 장착중
SIF TEQUIP:ARG:착유기
	CALLFORM EQUIP_COM{COM착유기}(ARG)

;눈가리개 장착중
SIF TEQUIP:ARG:눈가리개
	CALLFORM EQUIP_COM{COM눈가리개}(ARG)

;밧줄로 긴박중
SIF TEQUIP:ARG:밧줄
	CALLFORM EQUIP_COM{COM밧줄}(ARG)

;볼개그 장착중
SIF TEQUIP:ARG:볼개그
	CALLFORM EQUIP_COM{COM볼개그}(ARG)

;관장＋애널 플러그 장착중
SIF TEQUIP:ARG:관장
	CALLFORM EQUIP_COM{COM관장}(ARG)

;확장벌룬 장착중
SIF TEQUIP:ARG:확장벌룬
	CALLFORM EQUIP_COM{COM확장벌룬}(ARG)

;애널전극 장착중
SIF TEQUIP:ARG:애널전극
	CALLFORM EQUIP_COM{COM애널전극}(ARG)

;비디오촬영중
SIF ARG > 0 && TEQUIP:MASTER:비디오촬영
	CALLFORM EQUIP_COM{COM비디오촬영}(ARG)

;수치플레이중
;SIF TEQUIP:ARG:30
;	CALL EQUIP_COM186(ARG)

;목욕탕플레이중
SIF TEQUIP:ARG:목욕탕플레이
	CALLFORM EQUIP_COM{COM목욕탕플레이}(ARG)

;샤워 사용중
SIF TEQUIP:ARG:샤워
	CALLFORM EQUIP_COM{COM샤워}(ARG)

;새댁플레이중
SIF TEQUIP:ARG:새댁플레이
	CALLFORM EQUIP_COM{COM새댁플레이}(ARG)

;V삽입중
SIF TEQUIP:ARG:50 >= 0 && !SOURCE:ARG:쾌V
	CALL EQUIP_INCERT_V(ARG)

;A삽입중
SIF TEQUIP:ARG:51 >= 0 && !SOURCE:ARG:쾌A
	CALL EQUIP_INCERT_A(ARG)

;한우산을씀중
SIF TEQUIP:ARG:한우산을씀 && OUTROOF(CFLAG:ARG:현재위치) && ARG == TARGET
	CALLFORM EQUIP_COM{COM한우산을씀}(ARG)
RETURN 1


code:
;삽입 계속의 처리
@EQUIP_INCERT_V(ARG)
#DIM 간역

간역 = TEQUIP:ARG:50
;-------------------------------------------------
;행위자
;-------------------------------------------------
TCVAR:ARG:116 = 간역
;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
;TCVAR:4 사정쾌감강도

CSTR:간역:10 = @"/{ARG}/"
TCVAR:간역:사정장소 = 1

CSTR:ARG:10 += @"{TEQUIP:ARG:50}/"
SIF TCVAR:ARG:사정장소 == 0 || TCVAR:ARG:사정장소 >= 8
	TCVAR:ARG:사정장소 = 13

IF ARG == MASTER
	PRINTFORML ＜%CALLNAME:간역%로부터 V인서트＞

	TCVAR:간역:사정쾌감강도 = 850
	TCVAR:ARG:사정쾌감강도 = 500 + (100 * ABL:ARG:V감각)

	DOWNBASE:간역:체력 += 15
	DOWNBASE:간역:기력 += 20
	DOWNBASE:ARG:체력 += 10
	DOWNBASE:ARG:기력 += 20
	SOURCE:ARG:쾌V += 200
	SOURCE:간역:쾌C += 200
	SOURCE:간역:사랑 += 25
	SOURCE:간역:성행동 += 250
	SOURCE:간역:달성 += 25
	SOURCE:간역:정복 += 500
	SOURCE:간역:일탈 += 50

ELSE
	PRINTFORML ＜%CALLNAME:ARG%에 V인서트＞

	TCVAR:간역:사정쾌감강도 = 700 + TEQUIP:ARG:자궁 * 100
	TCVAR:ARG:사정쾌감강도 = 750
	DOWNBASE:간역:체력 += 15
	DOWNBASE:간역:기력 += 20
	;시간정지중
	IF FLAG:70
		SOURCE:ARG:쾌V += 150
		CALL 삽입계속V고통처리(ARG)

	ELSE
		DOWNBASE:ARG:체력 += 10
		DOWNBASE:ARG:기력 += 20
		SIF TEQUIP:ARG:체위 == 6
			SOURCE:ARG:고통 += 300
		SIF TALENT:간역:형상 == 5
			SOURCE:ARG:고통 += 200
		SOURCE:ARG:쾌V += 150
		SOURCE:ARG:사랑 += 75
		SOURCE:ARG:노출 += 25
		SIF TCVAR:ARG:억지로
			SOURCE:ARG:반감 += 1000
		;체위 보너스
		IF TEQUIP:ARG:체위 == CFLAG:ARG:좋아하는체위
			SOURCE:ARG:사랑 += 100
			SOURCE:ARG:공순 += 50
		ENDIF
		CALL 삽입계속V고통처리(ARG)
	ENDIF

	IF CFLAG:ARG:우후후 == 2
		SOURCE:ARG:쾌V += 150
		SOURCE:ARG:사랑 += 100
		TCVAR:간역:사정쾌감강도 += 150
	ENDIF
ENDIF

;-------------------------------------------------
;경험의 처리
;-------------------------------------------------
EXP:간역:삽입경험 ++
EXP:ARG:V경험 ++
SIF !FLAG:70 && !CFLAG:ARG:장난
EXP:ARG:V성교경험 ++
SIF TEQUIP:ARG:체위 == 6
	EXP:ARG:V확장경험 ++
SIF FLAG:70 == 1
	EXP:MASTER:시간경험 += 1
IF CFLAG:MASTER:장난 == 2
	EXP:MASTER:수면간경험 += 1
	EXP:ARG:수면간경험 += 1
ENDIF
RETURN 1

@EQUIP_INCERT_A(ARG)
#DIM 간역

간역 = TEQUIP:ARG:51
;-------------------------------------------------
;행위자
;-------------------------------------------------
TCVAR:ARG:116 = 간역
;TCVAR:0 사정 대상 플래그(비트 0=TARGET 1=PLAYER 2=MASTER 3=ASSI 4=촉수 5=그 외)
;TCVAR:1 예비 사정 대상 플래그(비트를 사용하고 싶지 않은 인용, CFLAG:100으로 번역됩니다)
;	(1=TARGET 2=PLAYER 3=MASTER 4=ASSI 5=TARGET+ASSI 6=MASTER+ASSI -1=촉수 -2=그 외)
;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
;TCVAR:4 사정쾌감강도
CSTR:간역:10 = @"/{ARG}/"
TCVAR:간역:사정장소 = 2

CSTR:ARG:10 += @"{TEQUIP:ARG:51}/"
SIF TCVAR:ARG:사정장소 == 0 || TCVAR:ARG:사정장소 >= 8
	TCVAR:ARG:사정장소 = 12

IF ARG == MASTER
	PRINTFORML ＜%CALLNAME:간역%로부터 A인서트＞

	TCVAR:간역:사정쾌감강도 = 850
	TCVAR:ARG:사정쾌감강도 = 500 + (100 * ABL:ARG:A감각)

	DOWNBASE:간역:체력 += 15
	DOWNBASE:간역:기력 += 20
	DOWNBASE:ARG:체력 += 20
	DOWNBASE:ARG:기력 += 20
	SOURCE:ARG:쾌A += 200
	SOURCE:간역:쾌C += 200
	SOURCE:간역:사랑 += 25
	SOURCE:간역:성행동 += 250
	SOURCE:간역:달성 += 25
	SOURCE:간역:정복 += 500
	SOURCE:간역:일탈 += 50

ELSE
	PRINTFORML ＜%CALLNAME:ARG%에 A인서트＞
	TCVAR:간역:사정쾌감강도 = 900
	TCVAR:ARG:사정쾌감강도 = 750
	DOWNBASE:간역:체력 += 15
	DOWNBASE:간역:기력 += 20
	;시간정지중
	IF FLAG:70
		SOURCE:ARG:쾌A += 200
		CALL 삽입계속A고통처리(ARG)

	ELSE
		DOWNBASE:ARG:체력 += 20
		DOWNBASE:ARG:기력 += 20
		SIF TALENT:간역:형상 == 5
			SOURCE:ARG:고통 += 300
		SOURCE:ARG:쾌A += 200
		SOURCE:ARG:사랑 += 50
		SOURCE:ARG:노출 += 30
		SOURCE:ARG:일탈 += 100
		SIF TEQUIP:ARG:체위 == 6
			SOURCE:ARG:고통 += 300
		;체위 보너스
		IF TEQUIP:ARG:체위 == CFLAG:ARG:좋아하는체위 - 5
			SOURCE:ARG:사랑 += 100
			SOURCE:ARG:공순 += 50
		ENDIF
		CALL 삽입계속A고통처리(ARG)
	ENDIF

	IF CFLAG:ARG:우후후 == 2
		SOURCE:ARG:쾌A += 150
		SOURCE:ARG:사랑 += 100
		TCVAR:간역:사정쾌감강도 += 150
	ENDIF
ENDIF
;-------------------------------------------------
;경험의 처리
;-------------------------------------------------
EXP:간역:삽입경험 ++
EXP:ARG:A경험 ++
SIF !FLAG:70 && !CFLAG:ARG:장난
	EXP:ARG:A성교경험 ++
SIF TEQUIP:ARG:체위 == 6
	EXP:ARG:A확장경험 ++
RETURN 1

@삽입계속V고통처리(ARG)
;EXP:V경험을 본다(처녀는 있을 수 없다)
LOCAL = 200
LOCAL:1 = 150
IF EXP:ARG:V경험 < EXPLV:2
	TIMES LOCAL , 0.80
	TIMES LOCAL:1 , 1.20
ELSEIF EXP:ARG:V경험 < EXPLV:3
	TIMES LOCAL , 0.40
ELSEIF EXP:ARG:V경험 < EXPLV:4
	TIMES LOCAL , 0.10
ELSEIF EXP:ARG:V경험 < EXPLV:5
	TIMES LOCAL , 0.00
ELSEIF EXP:ARG:V경험 >= EXPLV:5
	TIMES LOCAL , 0.00
ENDIF

;PALAM:윤활을 본다
IF PALAM:ARG:윤활 < PALAMLV:1
	TIMES LOCAL, 3.00
	TIMES LOCAL:1 , 3.00
ELSEIF PALAM:ARG:윤활 < PALAMLV:2
	TIMES LOCAL , 2.00
	TIMES LOCAL:1 , 2.50
ELSEIF PALAM:ARG:윤활 < PALAMLV:3
	TIMES LOCAL , 0.60
	TIMES LOCAL:1 , 1.50
ELSEIF PALAM:ARG:윤활 < PALAMLV:4
	TIMES LOCAL , 0.50
	TIMES LOCAL:1 , 1.00
ELSEIF PALAM:ARG:윤활 < PALAMLV:5
	TIMES LOCAL , 0.10
	TIMES LOCAL:1 , 0.50
ELSEIF PALAM:ARG:윤활 >= PALAMLV:5
	TIMES LOCAL , 0.02
	TIMES LOCAL:1 , 0.30
ENDIF

SOURCE:ARG:고통 += LOCAL

SIF !FLAG:70
	SOURCE:ARG:반감 += LOCAL:1

@삽입계속A고통처리(ARG)
LOCAL = 400
LOCAL:1 = 250

;EXP:A경험을 본다
IF EXP:ARG:A경험 < EXPLV:1
	TIMES LOCAL , 3.00
	TIMES LOCAL:1 , 3.00
ELSEIF EXP:ARG:A경험 < EXPLV:2
	TIMES LOCAL , 1.20
	TIMES LOCAL:1 , 2.00
ELSEIF EXP:ARG:A경험 < EXPLV:3
	TIMES LOCAL , 0.80
	TIMES LOCAL:1 , 1.50
ELSEIF EXP:ARG:A경험 < EXPLV:4
	TIMES LOCAL , 0.60
	TIMES LOCAL:1 , 1.20
ELSEIF EXP:ARG:A경험 < EXPLV:5
	TIMES LOCAL , 0.40
ELSEIF EXP:ARG:A경험 >= EXPLV:5
	TIMES LOCAL , 0.20
ENDIF

;PALAM:윤활을 본다
IF PALAM:ARG:윤활 < PALAMLV:1
	TIMES LOCAL, 3.00
	TIMES LOCAL:1 , 3.00
ELSEIF PALAM:ARG:윤활 < PALAMLV:2
	TIMES LOCAL , 2.00
	TIMES LOCAL:1 , 2.50
ELSEIF PALAM:ARG:윤활 < PALAMLV:3
	TIMES LOCAL , 0.60
	TIMES LOCAL:1 , 1.50
ELSEIF PALAM:ARG:윤활 < PALAMLV:4
	TIMES LOCAL , 0.50
	TIMES LOCAL:1 , 1.00
ELSEIF PALAM:ARG:윤활 < PALAMLV:5
	TIMES LOCAL , 0.30
	TIMES LOCAL:1 , 0.50
ELSEIF PALAM:ARG:윤활 >= PALAMLV:5
	TIMES LOCAL , 0.20
	TIMES LOCAL:1 , 0.30
ENDIF

SOURCE:ARG:고통 += LOCAL

SIF !FLAG:70 && !CFLAG:ARG:장난
	SOURCE:ARG:반감 += LOCAL:1

@삽입불가(ARG)
;-2 = 헐렁헐렁으로 기분 좋지 않은 -1 = 자극이 약한 1= 경험이 필요 2= 경험과 지식이 필요 3= 난쟁이오나홀
#FUNCTION
#DIM 당신의사이즈
#DIM 사이즈차이
IF ARG == PLAYER
	RETURNF 0
ENDIF

SELECTCASE TALENT:PLAYER:형상
	CASE 0,3,4
		당신의사이즈 = 0
	CASE 1
		당신의사이즈 = -1
	CASE 2
		당신의사이즈 = -2
	CASE 6
		당신의사이즈 = 1
	CASE 5,10
		당신의사이즈 = 2
ENDSELECT

사이즈차이 = (TALENT:PLAYER:체형 - TALENT:ARG:체형) * 2 + 당신의사이즈

SELECTCASE 사이즈차이
	CASE IS > 7
		RETURNF 3
	CASE 5 TO 7
		RETURNF 2
	CASE 3 TO 4
		RETURNF 1
	CASE -2 TO 2
		RETURNF 0
	CASE -3 TO -6
		RETURNF -1
	CASE IS < -6
		RETURNF -2
ENDSELECT


code:
;FileName_TRACHECK.ERB ----------------------------- Rev1. 00
;처녀상실시의 처리
;CALL		USER
;ARG		0:캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@LOST_VIRGIN(ARG)
#DIM DYNAMIC 상실자No
#DIMS DYNAMIC 상실자명
#DIM DYNAMIC 취득자No
#DIMS DYNAMIC 취득자명
#DIMS 수식어
;-------------------------------------------------
;처녀상실의 체크
;-------------------------------------------------
IF TALENT:ARG:처녀 == 1 && EXP_UP(1,ARG) > 0 && !TFLAG:2 && GETBIT(TALENT:ARG:성별,0) ;まずＶがないと！みんな有るけど…念の為。

	수식어 = ""
	IF ARG == MASTER
		CFLAG:ARG:처녀상실상대 = TARGET
		상실자No = MASTER
		상실자명 = %CALLNAME:MASTER%
		취득자No = MASTER_POSE(SET_P,SET_V)
		;P 이외로 상실했을 경우, MASTER_POSE(SET_P,SET_V)는 0이 되어, 결과적으로 취득자명까지 당신이 되어 버린다
		;따라서 취득자No=0이라면 TARGET를 대입
		SIF !취득자No
			취득자No = TARGET
		취득자명 = %CALLNAME:취득자No%
	ELSE
		CFLAG:ARG:처녀상실상대 = PLAYER
		상실자명 = %CALLNAME:ARG%
		상실자No = ARG
		취득자명 = %CALLNAME:MASTER%
		취득자No = MASTER
	ENDIF
	IF ARG != MASTER
		;정조 관념
		IF TALENT:ARG:정조 > 0
			IF !TALENT:ARG:연모
				TIMES SOURCE:ARG:사랑 , 0.60
				TIMES SOURCE:ARG:반감 , 10.00
			ENDIF
		;정조무관심
		ELSEIF TALENT:ARG:정조 < 0
			TIMES SOURCE:ARG:사랑 , 0.60
			TIMES SOURCE:ARG:반감 , 0.50
		ELSEIF TALENT:ARG:무지
			TIMES SOURCE:ARG:사랑 , 0.30
			TIMES SOURCE:ARG:반감 , 0.75
			SOURCE:ARG:공포 += 3000
		ENDIF
		IF CFLAG:ARG:우후후 == 2
			TIMES SOURCE:ARG:사랑 , 1.50
			TIMES SOURCE:ARG:반감 , 0.25
		ENDIF
	ENDIF
	SETCOLOR 247,171,166
	PRINTFORMW 【처녀상실】
	IF TEQUIP:ARG:50 == MASTER && GROUPMATCH(TALENT:MASTER:형상,5,6)
		PRINTFORMW %취득자명%의 흉악한 자지가, %상실자명%의 순결을 가차 없이 찢었다……
		TIMES SOURCE:ARG:고통, 3.00
		TIMES SOURCE:ARG:공포, 3.00
	ENDIF
	IF FLAG:70
		IF ARG == MASTER
			PRINTFORMW %조사처리(취득자명,"는")% 의식이 없는 채, \@ TALENT:ARG:스침 == 1 ?  만난지 얼마 안된# \@ %상실자명%의 처녀를 빼앗겼다……
		ELSE
			PRINTFORMW %조사처리(상실자명,"는")% 의식이 없는 채, \@ TALENT:ARG:스침 == 1 ?  만난지 얼마 안된# \@ %취득자명%에게 처녀를 빼앗겨 버렸다……
		ENDIF
	ELSE
		;함락 소질에 의한 처리
		;MASTER 상실
		IF ARG == MASTER
			IF TCVAR:TARGET:만취
				PRINTFORMW 혼탁 한 채의 %조사처리(취득자명,"는")%, %상실자명%의 처녀를 빼앗았다……
			;수면간, 네가 파파가 되어!
			ELSEIF CFLAG:TARGET:수면
				PRINTFORMW 잠들어 있는 %조사처리(취득자명,"는")%, %상실자명%에 억지로 처녀를 빼앗겨 버렸다
			;무지(쇼타)로 처녀상실이라든지 최고……아니, 재고
			ELSEIF TALENT:TARGET:무지
				;IF GROUPMATCH(TALENT:MASTER:형상, 5, 6)
				;무지한 상대의 자지가 들어가있는 경우
				IF TEQUIP:MASTER:50 == 취득자No
					PRINTFORMW 성 지식이 부족한 %조사처리(취득자명,"는")% 무엇을 하셨는지 모른채, \@ TALENT:취득자No:2 & 2 ? 강렬한 쾌감과 # \@ %조사처리(상실자명,"이")% 흘리는 피에 무서워하고 있다……
				;자지 이외
				ELSE
					PRINTFORMW 파과혈을 본 %조사처리(취득자명,"는")%, 자신이 %상실자명%에게 상처를 입혀 버렸다고 생각해 갈팡잘팡하고 있다
				ENDIF
				PRINTFORMW %조사처리(상실자명,"는")% 파과의 아픔을 참으면서 %조사처리(취득자명,"을")% 안심시켰다
			;소질 연인
			ELSEIF IS_연인(TARGET) || TALENT:TARGET:연모 > 1 ;KR판 오리지날 추가연인 지원용 IS_연인
				PRINTFORMW %조사처리(취득자명,"는")% 황홀해 하는 표정으로, %상실자명%의 은밀한 곳으로부터 흘러넘치는 선혈을 응시하고 있다……
				TFLAG:150 = 2
				TIMES SOURCE:TARGET:사랑 , 3.00
				TIMES SOURCE:TARGET:반감 , 0.10
			;연모
			ELSEIF TALENT:TARGET:연모
				PRINTFORMW %조사처리(취득자명,"는")% 기쁜듯한 미소를 띠며, 처녀를 바친 %조사처리(상실자명,"을")% 응시하고 있다……
				TFLAG:150 = 2
				TIMES SOURCE:TARGET:사랑 , 2.00
				TIMES SOURCE:TARGET:반감 , 0.30
				EXP:TARGET:애정경험 += 30
				CFLAG:ARG:처녀상실대상 |= 상실_연모
			ELSEIF TALENT:TARGET:사모
				IF TCVAR:TARGET:억지로 == 1
					PRINTFORMW %조사처리(취득자명,"는")% 쾌감에 당황하면서, %상실자명%에게 눈물 젖은 눈동자을 향하여 온다……
					TIMES SOURCE:TARGET:사랑 , 0.80
					TIMES SOURCE:TARGET:반감 , 1.50
					CFLAG:ARG:처녀상실대상 |= 상실_억지로
				ELSEIF TALENT:TARGET:정조 >= 0
					PRINTFORMW %조사처리(취득자명,"는")% 곤혹하면서, 어딘지 모르게 기쁜듯이 %조사처리(상실자명,"을")% 응시하고 있다……
					TIMES SOURCE:TARGET:사랑 , 1.30
					TIMES SOURCE:TARGET:반감 , 0.50
				ENDIF
				CFLAG:ARG:처녀상실대상 |= 상실_사모
			ELSE
				IF TCVAR:TARGET:억지로 == 1
					TIMES SOURCE:TARGET:사랑 , 0.10
					TIMES SOURCE:TARGET:반감 , 3.00
					PRINTFORMW %조사처리(취득자명,"는")%, 분노와 당혹감을 보이며 %조사처리(상실자명,"을")% 노려보았다……
					CFLAG:ARG:처녀상실대상 |= 상실_억지로
				ELSE
					PRINTFORMW %조사처리(취득자명,"는")% 무엇인가 말하고 싶어 하는 것 같은 표정을 지으며, %조사처리(상실자명,"과")%의 결합부를 응시하고 있다……
				ENDIF
			ENDIF
		;MASTER가 TARGET의 질에 삽입하고 있다
		ELSEIF TEQUIP:ARG:50 == MASTER
			IF TCVAR:ARG:만취
				PRINTFORMW %조사처리(상실자명,"는")% 혼탁 한 의식 속에서 처녀를 빼앗겨 버렸다……
			;수면간 좋은 아이는 자고 있는 동안에 마마로 해 버리자∼
			ELSEIF CFLAG:ARG:수면
				PRINTFORMW %상실자명%의 처음은 자고 있는 동안에 빼앗겨 버렸다……
			ELSEIF TALENT:ARG:무지
				PRINTFORMW 성 지식이 부족한 %조사처리(상실자명,"는")% 무엇을 하셨는지 모른채, 처녀상실의 아픔에 당황해 무서워하고 있다……
				PRINTFORMW %조사처리(취득자명,"는")% %조사처리(상실자명,"을")% 안심시켰다
			;친애 또한 상대가 주인(현재 친애가 없기 때문에, 연인으로 대용
			ELSEIF TALENT:ARG:연인 || TALENT:ARG:연모 > 1
				PRINTFORMW %조사처리(상실자명,"는")% %조사처리(취득자명,"아")% 처음의 상대인 일에 이 이상 없는 행복을 느끼고 있는 것 같다……
				TFLAG:150 = 2
				TIMES SOURCE:ARG:사랑 , 3.00
				TIMES SOURCE:ARG:반감 , 0.10
				 CFLAG:ARG:처녀상실대상 |= 상실_연모
			;연모 또한 상대가 주인
			ELSEIF TALENT:ARG:연모
				PRINTFORMW %조사처리(상실자명,"는")% %취득자명%에게 처녀를 바칠 수 있었던 일에 행복을 느끼고 있는 것 같다……
				TFLAG:150 = 2
				TIMES SOURCE:ARG:사랑 , 2.00
				TIMES SOURCE:ARG:반감 , 0.30
				EXP:ARG:애정경험 += 30
				CFLAG:ARG:처녀상실대상 |= 상실_연모
			ELSEIF TALENT:ARG:사모
				IF TCVAR:ARG:억지로 == 1
					PRINTFORMW 호의를 가지고 있던 상대에게 배신당한 %조사처리(상실자명,"는")%, 굵은 눈물을 흘리기 시작했다……
					TIMES SOURCE:ARG:사랑 , 0.80
					TIMES SOURCE:ARG:반감 , 1.50
					CFLAG:ARG:처녀상실대상 |= 상실_억지로
				ELSEIF TALENT:ARG:정조 >= 0
					PRINTFORMW 호의를 가지고 있었다고는 해도, %조사처리(상실자명,"는")% %취득자명%에게 정조까지도 허락해버린 것에 당혹해 하고 있다……
					TIMES SOURCE:ARG:사랑 , 1.30
					TIMES SOURCE:ARG:반감 , 0.80
				ENDIF
				CFLAG:ARG:처녀상실대상 |= 상실_사모
			ELSE
				IF TCVAR:ARG:억지로 == 1
					TIMES SOURCE:ARG:사랑 , 0.10
					TIMES SOURCE:ARG:반감 , 3.00
					IF CFLAG:ARG:패배자
						PRINTFORMW %조사처리(상실자명,"는")% 패배의 댓가로서, %취득자명%에게 처녀를 빼앗겨 버렸다……
					ELSE
						PRINTFORMW %상실자명%의 저항도 허무하게, %취득자명%에게 처녀를 빼앗겨 버렸다……
					ENDIF
					CFLAG:ARG:처녀상실대상 |= 상실_억지로
					IF TALENT:ARG:스침 == 1
						CSTR:ARG:모브코프로필2 = %REPLACE(@"%CSTR:ARG:모브코프로필2%","당해 지금은 얌전하다","당해, 처녀를 빼앗겼다")%
						CSTR:ARG:모브코프로필2 = %REPLACE(@"%CSTR:ARG:모브코프로필2%","애널처녀를 빼앗겼다","처녀를 빼앗겼다")%
					ENDIF
				ELSEIF TALENT:ARG:스침 == 1
					PRINTFORMW 비처로부터 흘러넘치는 피를 본 %조사처리(상실자명,"는")%, 만난지 얼마 되지 않은 %취득자명%에게 처녀를 바쳐 버린 것을 실감해 조금 후회했다……
				ELSE
					PRINTFORMW %조사처리(상실자명,"는")% 흘러가는 대로 %취득자명%에게 처녀를 빼앗겨 버렸다……
				ENDIF
			ENDIF
		ENDIF
	ENDIF
	IF CSTR:ARG:처녀상실이력 == ""
		IF GROUPMATCH(TCVAR:상실자No:카운터행동, 80, 83, 85)
			CALL SET_HISTORY_LOST_V(상실자No, 취득자No, "페니스")
		ELSEIF TCVAR:취득자No:카운터행동 == 57
			CALL SET_HISTORY_LOST_V(상실자No, 취득자No, "손가락")
		ELSEIF SELECTCOM == 44
			CALL SET_HISTORY_LOST_V(상실자No, 취득자No, "바이브")
		ELSE
			CALL SET_HISTORY_LOST_V(상실자No, 취득자No, "페니스")
		ENDIF
	ENDIF
ENDIF
RESETCOLOR
;-------------------------------------------------
;재생 처녀상실의 체크
;-------------------------------------------------
IF TALENT:ARG:처녀 == 2 && EXP_UP(1,ARG)
	TCVAR:ARG:파과 = 2
	;비디오촬영
	IF TEQUIP:28
		;성교
		IF TEQUIP:ARG:50
			LOCALS = LV1/
		ELSEIF TEQUIP:ARG:13
			LOCALS = LV3/
		;그 외
		ELSE
			LOCALS = LV4/
		ENDIF
		TSTR:1 += LOCALS
	ENDIF
	PRINTW 재생 처녀상실
ENDIF
;-------------------------------------------------
;무자각처녀상실의 체크
;-------------------------------------------------
IF TALENT:ARG:처녀 == -1 && EXP_UP(1,ARG) && CFLAG:ARG:292 == 1 && !FLAG:70 && !TCVAR:ARG:만취 && !CFLAG:MASTER:장난
	TCVAR:ARG:파과 = -1
	CFLAG:ARG:292 = 0
	PRINTW 처녀상실?
	SETCOLOR 247,171,166
	PRINTFORMW %ARG는% 고통인것 같은 고통을 느끼지 않아 곤혹스러워하고 있다…
	RESETCOLOR
ENDIF

@동정키스미경험상실(ARG)
#DIM 시추에이션
;시추에이션	 0:통상
;					 1:데이트 돌아오는 길
;					 2:딸딸이들킴
;					 4:페니스에
;					 8:자발적으로
;					16:무자각
;					32:매춘
시추에이션 = 0
VARSET LOCALS
FONTBOLD
SETCOLOR 255,100,255
IF !TALENT:ARG:비동정 && TALENT:ARG:성별 & 2
	IF TCVAR:ARG:동정상실 == 1
		PRINTFORMW 동정상실(%CALLNAME:ARG%)
		TALENT:ARG:비동정 = 1
	ENDIF
ENDIF

;키스미경험 상실
IF ARG == MASTER && TALENT:ARG:키스미경험 && GROUPMATCH(SELECTCOM,2,20,312)
	시추에이션 |= 8
	IF GROUPMATCH(SELECTCOM,20,312) && (TFLAG:193 != 99 && PREVCOM != 309)
		CALL SET_FIRSTKISS(MASTER, ARG, 시추에이션)
	ELSEIF GROUPMATCH(SELECTCOM,2) && TALENT:TARGET:성별 & 2
		시추에이션 |= 4
		CALL SET_FIRSTKISS(ARG, TARGET, 시추에이션)
	ENDIF
ELSEIF TALENT:MASTER:키스미경험 && GROUPMATCH(TCVAR:ARG:카운터행동,30,39,60,71,91)
	IF GROUPMATCH(TCVAR:ARG:카운터행동,30,39,60)
		CALL SET_FIRSTKISS(MASTER, ARG, 시추에이션)
	ELSEIF GROUPMATCH(TCVAR:ARG:카운터행동,71,91) && TALENT:ARG:성별 & 2
		시추에이션 |= 4
		CALL SET_FIRSTKISS(MASTER, ARG, 시추에이션)
	ENDIF
ENDIF
IF TALENT:ARG:키스미경험 && (TCVAR:ARG:FK상실 || GROUPMATCH(TCVAR:ARG:카운터행동,30,39,60,51))
	;무자각
	SIF CFLAG:ARG:수면 || FLAG:70
		시추에이션 |= 16
	;자발적으로
	SIF GROUPMATCH(TCVAR:ARG:카운터행동,30,39,60,51)
		시추에이션 |= 8
	;페니스에
	SIF TCVAR:ARG:FK상실 == 2 || GROUPMATCH(TCVAR:ARG:카운터행동,51)
		시추에이션 |= 4

	IF GROUPMATCH(SELECTCOM,20,312) && (TFLAG:193 != 99 && PREVCOM != 309)
		;이마 쪽을 제외한다
		CALL SET_FIRSTKISS(ARG, MASTER, 시추에이션)
	ELSEIF GROUPMATCH(SELECTCOM,81,88,140,337) || GROUPMATCH(TCVAR:ARG:카운터행동,30,60,51)
		CALL SET_FIRSTKISS(ARG, MASTER, 시추에이션)
	ELSEIF TCVAR:ARG:카운터행동 == 39
		CALL SET_FIRSTKISS(ARG, MASTER, 시추에이션, "엿과 함께")
	ENDIF
ENDIF

;애널 처녀상실
IF CFLAG:ARG:애널처녀상실대상 & 상실_시간정지 || !CFLAG:ARG:애널처녀상실기념일
	IF TCVAR:ARG:A상실발생
		;시간정지중에 경험이 끝난 상태
		IF !FLAG:70 && ARG && CFLAG:ARG:애널처녀상실대상 & 상실_시간정지
			SIF FIRSTTIME("A파과",0)
				PRINTFORMW 애널 처녀상실? (%CALLNAME:ARG%)
			TCVAR:ARG:A파과 = -1
		;완전하게 처음
		ELSEIF !CFLAG:ARG:애널처녀상실기념일
			PRINTFORMW 애널 처녀상실(%CALLNAME:ARG%)
			IF CSTR:ARG:모브코프로필2 != ""
				CSTR:ARG:모브코프로필2 = %REPLACE(@"%CSTR:ARG:모브코프로필2%","당해 지금은 얌전하다","당해 애널처녀를 빼앗겼다")%
			ENDIF
			IF SELECTCOM == 45
				LOCALS = 애널바이브
			ELSEIF SELECTCOM == 46
				LOCALS = 애널비즈
			ELSE
				LOCALS = 페니스
			ENDIF
			IF !FLAG:70
				TCVAR:ARG:A파과 = 1
			ELSE
				TCVAR:ARG:A파과 = -2
			ENDIF
			IF ARG == TARGET
				CALL SET_HISTORY_LOST_A(ARG, PLAYER, @"%LOCALS%")
			ELSE
				CALL SET_HISTORY_LOST_A(ARG, TARGET, @"%LOCALS%")
			ENDIF

		ENDIF
	ENDIF
ENDIF
FONTREGULAR
RESETCOLOR

@LOST_VIRGIN_STOP(ARG)
SIF !SHIRAHU(ARG)
	RETURN
IF CFLAG:ARG:파과캔슬구상있음
	;말측으로부터 간섭 가능하게
	CALL KOJO_MESSAGE_SEND("LOST_VIRGIN_STOP", 0, ARG)
	IF RESULT == 1
		RETURN -1
	ELSE
		RETURN 0
	ENDIF
	RESETCOLOR
ELSEIF !TALENT:ARG:무지 && !TALENT:ARG:연모 && !TALENT:ARG:애욕 && !TALENT:ARG:연인 && !TCVAR:ARG:기다리지않다 && !CFLAG:장난
;정조 관념 소유 or정조 보통 또한 사모 없음
	IF TALENT:ARG:정조 > 0 || (!TALENT:ARG:정조 && !TALENT:ARG:사모)
		PRINTFORMW %ARG는% 그건 기다렸으면 좋겠다고 말하기 시작했다
		IF (EXP:ARG:60 + EXP:ARG:61 + EXP:ARG:62 + EXP:ARG:63 + EXP:ARG:64) == 0
			IF TALENT:ARG:정조 > 0
				PRINTFORMW 과연 몸가짐이 단단하다……
			ELSE
				PRINTFORMW 의외로 가드가 단단하다……
			ENDIF
		ELSE
			PRINTFORMW 귀찮은 아가씨다……
		ENDIF
		CALL ASK_YN("포기한다","차린 밥상을 먹지 않는 건")
		IF RESULT == 1
			IF TALENT:ARG:정조 > 0 && ABL:ARG:전투능력 > ABL:MASTER:전투능력 && !TCVAR:ARG:억지로
				PRINTFORMW %마스터는% 억지로 강요했지만, %ARG가% 너무나 격렬하게 저항해 포기했다
					RETURN 1
			ELSEIF TALENT:ARG:정조 > 0
				PRINTFORMW %ARG는% 눈에 눈물을 머금어 %마스터를% 거절했다
				PRINTL [0] 진심으로 싫을 것 같은 것으로 포기한다
				PRINTL [1] 해 버린다
				INPUT
				IF !RESULT
					PRINTFORMW %ARG는% 안심하고 있다…
					PRINTFORMW %마스터는% 조금 상처받았다
					RETURN -1
				ELSE
					PRINTFORMW %마스터는% 싫어하는 %ARG를% 덮쳤다…
					TCVAR:ARG:기다리지않다 = 1
					TCVAR:ARG:억지로 = 1
					RETURN 0
				ENDIF
			ELSE
				PRINTFORMW %마스터는% 억지로 밀어붙였다…
				TCVAR:ARG:기다리지않다 = 1
				RETURN 0
			ENDIF
		ELSE
			PRINTFORMW %ARG는% 안심하고 있다…
			RETURN -1
		ENDIF
	ENDIF
ENDIF

@SET_FIRSTKISS(ARG, 키스상대, 시추에이션, 수식어)
#DIM 키스상대
;시추에이션	 0:통상
;					 1:데이트 돌아오는 길
;					 2:딸딸이들킴
;					 4:페니스에
;					 8:자발적으로
;					16:무자각
;					32:매춘
#DIM 억지로시츄
#DIM 시추에이션
#DIMS 수식어
#DIM 스스로

VARSET LOCALS
스스로 = 0
억지로시츄 = (TCVAR:ARG:억지로 && !(TALENT:ARG:애욕 || TALENT:ARG:섹프 || TALENT:ARG:연모 || TALENT:ARG:사모))
;억지로状態で、キス合意はつかない。억지로だから！！！唇を許すわけ無いでしょ！！　ただし、陥落素質があればギリギリセーフ。

;キス経験済み
IF TALENT:ARG:키스미경험 == 0
	RETURN
;無自覚キス経験済み
ELSEIF TALENT:ARG:키스미경험 < 0
	;無自覚キス同士では重複しない
	SIF 시추에이션 & 16
		RETURN
ENDIF

;FLAG:FK판정에 따라서는 돌아간다
SIF !(FLAG:FK판정 & 1) && 시추에이션 & 16
	RETURN
SIF !(FLAG:FK판정 & 2) && 시추에이션 & 4
	RETURN

IF ARG == 키스상대
	IF ARG == MASTER
		키스상대 = TARGET
	ELSE
		키스상대 = MASTER
	ENDIF
ENDIF

SETCOLOR 247,171,166
PRINTFORM 키스미경험 상실
SIF 시추에이션 & 16
	PRINT ?
PRINTFORMW (%CALLNAME:ARG%)
RESETCOLOR

CFLAG:ARG:첫키스상실기념일 = DAY
TALENT:ARG:키스미경험 = 0
CFLAG:ARG:첫키스상실상대 = 키스상대

;대상과 수식어 결정
IF TALENT:ARG:사모
	스스로 = 1
	LOCALS:0 = 마음에 두고 있었던
ELSEIF ARG == MASTER && IS_연인(키스상대) ;KR판 오리지날 추가연인수 지원용 IS_연인
	스스로 = 1
	LOCALS:0 = 사랑하는
ELSEIF TALENT:ARG:연인 || TALENT:ARG:연모
	스스로 = 1
	LOCALS:0 = 사랑하는
ENDIF
IF 시추에이션 & 2
	IF ARG != MASTER
		LOCALS:2 = 자위중에, 침입해온
	ELSE
		LOCALS:2 = 자위하고 있던
		스스로 = 1
	ENDIF
ENDIF
IF 시추에이션 & 4
	LOCALS:1 = 의 페니스
	CFLAG:ARG:첫키스상실대상 = 1
ENDIF
SIF 시추에이션 & 8
	스스로 = 1
IF 시추에이션 & 16
	IF FLAG:70
		LOCALS:2 = 시간정지중에
	ELSEIF CFLAG:ARG:수면
		LOCALS:2 = 자고 있는 동안에
	ELSE
		LOCALS:2 = 자신도 모르는 사이에
	ENDIF
	CFLAG:ARG:첫키스상실대상 = 2
	TALENT:ARG:키스미경험 = -1
ENDIF
SIF TALENT:ARG:무지 && 시추에이션 & 4
	LOCALS:2 = %LOCALS:2% 그 의미도 알지 못한 채

SIF 억지로시츄
	LOCALS:2 = %LOCALS:2% 억지로

;장소 결정
IF 시추에이션 & 1
	CSTR:ARG:첫키스위치 = 데이트하고 돌아오는 길
	SIF ARG != MASTER
		스스로 = 1
ELSEIF 시추에이션 & 2
	IF ARG != MASTER
		CSTR:ARG:첫키스위치 = 자기 방
	ELSE
		CSTR:ARG:첫키스위치 = %CALLNAME:키스상대% 댁
	ENDIF
ELSE
	IF CHK_DATENOW(CFLAG:ARG:데이트중) && CFLAG:ARG:현재위치 != OMANEKIBEYA()
		CSTR:ARG:첫키스위치 = %DATENAME_SPOT()%
	ELSEIF CHK_DATENOW(CFLAG:MASTER:데이트중) && CFLAG:MASTER:현재위치 != OMANEKIBEYA()
		CSTR:ARG:첫키스위치 = %DATENAME_PLACE()%의 %DATENAME_SPOT()%
	ELSE
		CSTR:ARG:첫키스위치 = %NAME_FROM_PLACE(CFLAG:ARG:현재위치)%
	ENDIF
ENDIF

;동사 결정
IF 수식어 == "엿과 함께"
	LOCALS:1 += "와"
	LOCALS:3 = 주고 받았다
ELSEIF 스스로
	LOCALS:1 += "에게"
	LOCALS:3 = 바쳤다
ELSEIF 시추에이션 & 32
	LOCALS:1 += "에게"
	LOCALS:3 = 빼앗아 주었다
ELSE
	LOCALS:1 += "에게"
	LOCALS:3 = 빼앗겼다
ENDIF

; 개인 임의수정, 띄어쓰기 추가
;---- EDIT START ---------------------------------
SIF LOCALS:0 != ""
	LOCALS:0 += " "
SIF LOCALS:1 != "" && 수식어 == "엿과 함께"
	LOCALS:1 += " "
SIF LOCALS:2 != ""
	LOCALS:2 += " "
;---- EDIT END -----------------------------------

IF CSTR:ARG:첫키스상실이력 == ""
	CSTR:ARG:첫키스상실이력 = %PRINT_DATE_F(DAY)%, %CSTR:ARG:첫키스위치%에서 %LOCALS:2%%LOCALS:0%%CALLNAME:키스상대%%LOCALS:1%%수식어% 첫키스를 %LOCALS:3%
ELSE
	;무자각키스가 있는 경우에만 중복된다
	CSTR:ARG:첫키스상실이력 = %PRINT_DATE_F(DAY)%, %CSTR:ARG:첫키스위치%에서 %LOCALS:2%%LOCALS:0%%CALLNAME:키스상대%%LOCALS:1%%수식어% 첫키스를 %LOCALS:3%\BR%CSTR:ARG:첫키스상실이력%
ENDIF
;기정사실도 붙여 둔다
SIF ARG != MASTER && !(시추에이션 & 16) && !억지로시츄
	SETBIT CFLAG:ARG:기정사실, 0
;애정경험도 여기서 들어갈 수 있기로 한다
SIF ARG != MASTER && !(시추에이션 & 16) && !억지로시츄
	EXP:ARG:애정경험 += 5

;===================================================================================
;이력 캐릭터라인 설정용 함수군
;대상이 미기입이라면"페니스"가 대입됩니다.
;또 그 경우 포함 대상이 페니스"라면 상대가 나지 않은 경우 자동으로 페니스밴드가 됩니다.
;장소가 미기입이라고 자동으로 위치를 취득합니다.
;시간정지중이라고 만취하고 있는이라고는 자동으로 대입됩니다.
;각종 플래그 처리도 함수측에서 설정합니다.
;즉 호출해 옆에서 분기 들어갈 수 있을 필요가 거의 없어졌습니다.
;ARG가 당신의 페니스로 처녀상실 했다면 극단적인 이야기 CALL SET_HISTORY_LOST_V(ARG)만으로 OK.
;잘되었네. 굉장하다.
;===================================================================================

@SET_HISTORY_LOST_A(ARG, 상대, 대상, 장소, 수식어)
#DIM 상대
#DIMS 대상
#DIMS 장소
#DIMS 수식어

CFLAG:ARG:애널처녀상실기념일 = DAY
CFLAG:ARG:애널처녀상실상대 = 상대

CSTR:ARG:애널처녀상실이력 = %HISTORY_LOST_TEXT(ARG, 상대, 대상, 장소, 수식어,"항문 처녀")%

@SET_HISTORY_LOST_V(ARG, 상대, 대상, 장소, 수식어)
#DIM 상대
#DIMS 대상
#DIMS 장소
#DIMS 수식어

;비처녀화, 처녀식 플래그, 상실 기념일등도 함수측에서 설정
SIF ARG != MASTER
	FLAG:처녀식 ++
CFLAG:ARG:처녀상실기념일 = DAY
TALENT:ARG:처녀 = 0
TCVAR:ARG:파과 = 1
CFLAG:ARG:처녀상실상대 = 상대
CSTR:ARG:처녀상실이력 = %HISTORY_LOST_TEXT(ARG, 상대, 대상, 장소, 수식어, "처녀")%

@HISTORY_LOST_TEXT(ARG, 상대, 대상, 장소, 수식어, 상실한것)
#FUNCTIONS
#DIM 상대
#DIMS 대상
#DIMS 장소
#DIMS 수식어
#DIMS 상실한것
#DIMS 상황
#DIM 스스로

VARSET LOCALS
스스로 = 0
IF ARG == MASTER || (CFLAG:MASTER:우후후 != 2 && ARG != MASTER && ARG != TARGET)
	LOCALS:0 =
ELSEIF TALENT:ARG:사모
	스스로 = 1
	LOCALS:0 = 마음에 두고 있던
ELSEIF TALENT:ARG:연인 || TALENT:ARG:연모
	스스로 = 1
	LOCALS:0 = 사랑하는
ELSEIF !CFLAG:ARG:안면 && ARG != MASTER
	LOCALS:0 = 보지 않고 모르는
ELSEIF TALENT:ARG:스침 == 1
	LOCALS:0 = 초대면의
ENDIF

SIF !SHIRAHU(ARG)
	스스로 = 0
SIF 대상 == ""
	대상 = 페니스

IF FLAG:70
	상황 = 시간정지중에
ELSEIF TCVAR:ARG:만취
	상황 = 만취하고 있을 때
ELSEIF CFLAG:ARG:수면
	상황 = 자고 있는 동안에
ELSEIF ARG != MASTER && ARG != TARGET && CFLAG:MASTER:우후후 != 2
	상황 = %CALLNAME:MASTER%의 소망대로
ELSE
	상황 =
ENDIF

SIF ARG == MASTER && TALENT:상대:무지
	상황 += "아무것도 모르는 "

IF 대상 == "페니스"
	IF TALENT:상대:2 & 2
		LOCALS:1 = %CALLNAME:상대%의 %TINKO_SIMPLE(상대)%에
	ELSE
		LOCALS:1 = %CALLNAME:상대%에 페니스밴드로
	ENDIF
ELSEIF 대상 == "손가락"
	LOCALS:1 = %CALLNAME:상대%의 손가락에
ELSE
	LOCALS:1 = %CALLNAME:상대%에 %대상%로
ENDIF
SIF 장소 == ""
	장소 = %PLACE_NAME(ARG)%
IF 수식어 == ""  && SHIRAHU(ARG)
	IF TCVAR:ARG:억지로 && CFLAG:ARG:패배자
		수식어 = 패배의 댓가로서
		스스로 = 0
	ELSEIF TCVAR:ARG:억지로
		수식어 = 힘으로
		스스로 = 0
	ELSEIF TALENT:ARG:무지
		수식어 = 아무것도 모르는 채
		스스로 = 0
	ELSEIF ARG != MASTER && ARG != TARGET && CFLAG:MASTER:우후후 != 2
		수식어 = 불만을 느끼면서
	ELSEIF TALENT:ARG:연인
		수식어 = 행복감 속에서
	ELSEIF CFLAG:ARG:우후후 == 2
		수식어 = 스스로
		스스로 = 1
	ENDIF
ENDIF

IF 스스로
	LOCALS:3 = 바쳤다
ELSEIF ARG == MASTER || 수식어 == "치료라고 칭하며"
	LOCALS:3 = 주었다
ELSEIF 수식어 == "돈을 내고"
	LOCALS:3 = 빼앗아 주었다
ELSE
	LOCALS:3 = 빼앗겼다
ENDIF

; 개인 임의수정, 띄어쓰기 추가
;---- EDIT START ---------------------------------
SIF 상황 != ""
	상황 += " "
SIF LOCALS:0 != ""
	LOCALS:0 += " "
SIF LOCALS:1 != ""
	LOCALS:1 += " "
;---- EDIT END -----------------------------------

LOCALS:10 = %PRINT_DATE_F(DAY)%, %상황%%장소%에서 %LOCALS:0%%LOCALS:1%%수식어% %상실한것%를 %LOCALS:3%

RETURNF LOCALS:10

;당신의 경우 이벤트로 CSV로 정의되어 있지 않은 상대와 하는 일도 있을 수 있으므로보다 자유도의 높은 것을
@SET_FIRSTKISS_MASTER(ARGS)
IF TALENT:MASTER:키스미경험
	CALL COLORMESSAGE(@"키스미경험 상실(%CALLNAME:MASTER%)", C_D_PINK, 2)
	CFLAG:MASTER:첫키스상실기념일 = DAY
	TALENT:MASTER:키스미경험 = 0
	CSTR:MASTER:첫키스상실이력 = %PRINT_DATE_F(DAY)%, %ARGS%
ENDIF